EdicinWing 編集スクリプト CookBook

for Version 5.0

Contents

● 基本書式

1	変数 ・・・・・・・・・・
2	$\neg \forall \forall \forall \land $
3	コンソールに出力する ・・・・・・・・・・・・・・4
4	他の編集スクリプトを読み込む ・・・・・・・・・5
5	手続きを定義する ・・・・・・・・・・・・・・・・・・・・6
6	条件分岐 ・・・・・・・・・・・・・・・・・・・・・7
7	繰り返し ・・・・・・・・・・・・・・・・・8
8	スクリプトを終了させる ・・・・・・・・・・・・・9
9	エラー終了 ・・・・・・・・・・・・・・・10
10	論理演算 · · · · · · · · · · · · · · · · · · ·
11	比較演算子
12	数値演算 ・・・・・・・・・・・・・・・・・・・・13
13	データスタックの操作コマンド・・・・・・・・・・14
14	配列、辞書を作成する ・・・・・・・・・・・・16
15	配列、辞書に要素を追加する · · · · · · · · · · · · · · · 17
16	配列、辞書の要素を取得する · · · · · · · · · · · · · · · 18
17	配列、辞書の要素数を取得する ・・・・・・・・・・19
18	文字列を連結する · · · · · · · · · · · · · · · · · · ·
19	文字列の長さを調べる ・・・・・・・・・・・・21
20	文字列を比較する ・・・・・・・・・・・・22
21	文字列を部分的に比較する ・・・・・・・・・・・23
22	文字列を検索する $\cdot \cdot \cdot$
23	文字列の一部を取り出す ・・・・・・・・・・・25
24	数値を文字列に変換する ・・・・・・・・・・26
25	整数型へ変換する ······27
26	実数型へ変換する ・・・・・・・・・・・・・28
27	浮動小数点数を丸める ・・・・・・・・・・・29

28	オブジェクトの型を調べる・・・・・・・・・・・30
29	ファイルを作成する ・・・・・・・・・・・・・31
30	ファイルを閉じる・・・・・・・・・32
31	ファイルへ文字列を書き込む・・・・・・・・・・・33
32	改行コードを使う ・・・・・・・・・・・・・・・34
33	ファイルが存在しているかチェックする ・・・・・・・・・35
●編	集コマンド
34	
35	枠の選択を解除する・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・
36	枠選択状態を取得する ・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・
37	連結した小組枠を選択する・・・・・・・・・・・・・・・SS
38	枠のサイズを取得する ・・・・・・・・・・・・・・・・・・4C
39	枠の位置を取得する・・・・・・・・・・・・・・・・41
40	枠の属性を変更する・・・・・・・・・・・・・・・・・42
41	選択枠を「部品書出」する ・・・・・・・・・・・・44
42	選択枠を EPS 画像にする · · · · · · · · · · · · · · · 45
43	選択枠のプレビュー画像を作成する・・・・・・・・・・47
44	固定枠を作成する・・・・・・・・・・・・・・・・49
45	枠種別の取得する · · · · · · · · · · · · · · · · · · ·
46	枠名称の一覧を取得する ・・・・・・・・・・51
47	枠の余白を取得する・・・・・・・・・・・・・・・53
48	選択中の枠から次の枠を選択する・・・・・・・・・54
49	カットバッファを変更する ・・・・・・・・・・55
50	選択中の枠を削除する・・・・・・・・・・・・・・56
51	選択中の枠をカットする・・・・・・・・・・・57
52	コピーした枠をペーストする ・・・・・・・・・58
53	オブジェクトを全選択する・・・・・・・・・・・55
54	基準点を設定する・・・・・・・・・・・・・・・・6C
55	基準点に揃える・・・・・・・・・・・・・・・・・61
56	基準点の位置を調べる・・・・・・・・・・・・・・・62

5	7	ページを変更する ・・・・・・・・・・・・・・・63
5	8	現在ページを調べる・・・・・・・・・・・・・・・・64
5	9	画像を張り込む・・・・・・・・・・・・・・・・・65
6	0	画像領域を作成する・・・・・・・・・・・・・・・・66
6	31	画像を入れ替える・・・・・・・・・・・・・・・67
6	2	画像をファイルとして取り出す ・・・・・・・・・・・・・・68
6	3	選択している枠のあふれ状態を調べる・・・・・・・・・・69
6	4	記事全体のあふれ状態を調べる・・・・・・・・・・・・・70
6	5	あふれ情報をファイルに書き出す ・・・・・・・・・・・71
6	6	選択している枠の残り行数を取得する ・・・・・・・・・72
6	57	空領域を取得する ・・・・・・・・・・・・・・・73
6	8	文字が読み込まれている領域を調べる・・・・・・・・・75
6	9	あふれている枠を拡張する ・・・・・・・・・・・76
7	O'	行間を調整して追込む ・・・・・・・・・・・・・・77
7	1	カラーパレットに色を追加する ・・・・・・・・・・・78
7	2	カラーパレットの色の成分を変更する ・・・・・・・・・・79
7	'3	カラーパレットの色を削除する・・・・・・・・・・80
7	'4	オブジェクトの色を変更する・・・・・・・・・・・81
7	'5	カラーパレットに色が存在するかチェックする ・・・・・・・・82
7	'6	トリガーテキストを読み込む ・・・・・・・・・・・83
7	7	トリガーテキストを取得する ・・・・・・・・・・・85
7	'8	選択している文字列のトリガーテキストを文字列で取得する ・・・・・86
7	'9	トリガー書出オプション定義ファイルを作成する ・・・・・・・87
8	80	セルの選択をする ・・・・・・・・・・・・・90
8	31	セル属性をコピー・ペーストする ・・・・・・・・・・93
8	32	選択中のセルの文字列の選択する ・・・・・・・・・・94
8	3	セルの選択位置を変更する ・・・・・・・・・・・・95
8	84	選択中のセルの位置情報を取得する ・・・・・・・・・・96
8	35	セルを結合する・・・・・・・・・・・・・・・97
8	86	セルの分割する ・・・・・・・・・・・・・・・98
8	37	選択中のセルの属性を変更する ・・・・・・・・・・99
8	88	選択中の表のセルサイズを調整する・・・・・・・・・・101

89	選択中のセルから条件にあった次のセルを選択する・・・・・・・・1	02
90	選択中の文字列を削除する・・・・・・・・・・・・・・・	05
91	文字列の選択状態を調べる・・・・・・・・・・・・・・	06
92	文字列の選択を解除する・・・・・・・・・・・・・・・・	07
93	単位付き数値文字列の単位を変換する・・・・・・・・・・・	80
94	記事名を取得する・・・・・・・・・・・・・・・・・・・	
95	文書名を取得する・・・・・・・・・・・・・・・・・・・	
96	文章内データ一覧を取得する・・・・・・・・・・・・・	11
97	差込情報の一覧を取得する ・・・・・・・・・]	13
98	ノンブルからページ番号へ変換する・・・・・・・・・・・1	
99	ページ番号からノンブルへ変換する・・・・・・・・・・・1	16
100	環境変数の値の取得する・・・・・・・・・・・・・・・)	17
• +-	L=T > . 19	
● 刈	話コマンド	
101	警告パネルにメッセージを表示させる・・・・・・・・・]	19
102	確認パネルにメッセージを表示させる・・・・・・・・・)	20
103	選択パネルを使用する・・・・・・・・・・・・・・・・・	
104	選択中の枠の枠名称を取得する・・・・・・・・・・・1	22

変数

ver. 5.0.0.0 以降

●宣言

変数の宣言には var を使用します。

var <変数名>

●オブジェクトの型

編集スクリプトで使用できるオブジェクトの型には以下のものがあります。

文字列型	""で囲まれた文字列 "自身は¥(0x5c)でエスケープする / で始まる文字列でもよい	"string" "あいうえお" /リテラル文字列
整数型	0~9 が並んだ文字列で、その先頭に「-」があってもよい	0 1 2 3 -1 -2 -3
実数型	整数型の文字列に加えて「.」がどこかに 1 つある文字列	1. 23456 -1. 234567
配列型	1 つ以上のオブジェクトを、[と] で囲んだもの 任意の型のオブジェクトを配列の要素に出来る	[12345]
辞書型	1つ以上の辞書要素を、<<と>>で囲んだもの 辞書要素とは文字列型と任意の型のオブジェクトとのペ アのことを指す	<< /a 1 /b 2 >>
ファイル型	open コマンドの返り値として取得されるオブジェクト	
単位付き数 値文字列型	文字列型のうち、数値+単位記号の形式で表記されたもの 使用できる単位は"mm"、"pt"、"ees"、"Q"、"H"、"%" 1ees=1/2400inch (1inch=25.4mm) オブジェクト自体は文字列型として扱われる	"1.0001ees"

●代入

変数に値(データオブジェクト)を代入するときは、@代入コマンドを使用します。

コマンド名:@代入

パラメータ:変数名 代入するデータ(任意のデータオブジェクト)

返り値 : なし

●参照

変数の値を参照するときは、0参照コマンドを使用します。

参照とは値(データオブジェクト)をデータスタックに積むことを表します。

コマンド名:@参照

パラメータ:変数名

返り値 :変数の値

サンプルコード

var n // 変数の宣言

1 n @代入 // 変数 n に 1 を代入

n @参照 // データスタックに 1 が積まれる

コメント

ver. 5.0.0.0 以降

スクリプト中にスクリプトではない文字列を埋め込むには、コメントを使用します。 スクリプト中「//」で始まる行文字列は以降、改行までコメントとなります。 //の後ろには1個以上のスペースが必要です。

サンプルコード

// コメント

23add // コメント

コンソールに出力する

ver. 5.0.0.0 以降

スタックの先頭の内容をコンソール(edicons 及び編集スクリプトログ)に出力すには「.」を使用します。

コマンド名:.

パラメータ:任意のデータオブジェクト

返り値 : なし

サンプルコード

"ログへ出力" . //「ログへ出力」をコンソールに出力

23 add dup . // スタックの先頭の内容を複製してコンソールに出力

他の編集スクリプトを読み込む

ver. 5.0.0.0 以降

他の編集スクリプトを読み込むには include を使用します。

include <ファイルパス>

<ファイルパス>には、編集スクリプトのファイル名を文字列型で指定します。

<ファイルパス>が、「/」で始まっているときは絶対パス、「./」で始まっているときはinclude 元からの相対パス、それ以外のときは環境変数での設定値からの相対パスとなります。インクルードファイルパス用環境変数 EDI_ESL_INCLUDE が設定されていない場合はエラーとなります。

サンプルコード

include "/@c/EDIAN10A/EDIDATA/ediesI/mod.esI" // 絶対パス include "./mod.esI" // include 元からの相対パス

include "mod.esl" // 環境変数 EDI_ESL_INCLUDE からの相対パス

手続きを定義する

ver. 5.0.0.0 以降

ユーザ定義の手続きのを作成するには以下の構文を使用します。

```
: <手続き名>
<処理>
;
```

「:」と<手続き名>の間には1個以上のスペースが必要です。

```
// @代入の代わりをする set を定義
: set
@代入
;
var n
6 n set //nに6を代入
exit
```

条件分岐

ver. 5.0.0.0 以降

条件分岐には if 文を使用します。

<条件> if <真の場合の処理> else <偽の場合の処理> endif

<条件>のところは真偽値を返すコマンドを使用し、<真の場合の処理>、<偽の場合の処理>のところには、処理したいコマンドの並びを記述してください。

else <偽の場合の処理>の部分は省略することが出来ます。

```
//nが6だったら「true」、6以外だったら「false」をコンソールに表示
n @参照 6 == if
"true" .
else
"false" .
endif
```

繰り返し

ver. 5.0.0.0 以降

繰り返しには loop 文を使用します。

(<条件>) <ループの中の処理> loop

<条件>のところは、真偽値を返すコマンドを使い、<条件>が偽になったときに、このループから抜けます。

<ループの中の処理>のところは、ループ中に処理したいコマンドの並びを記述します。 ループの中の処理時にループから抜けたい場合には break コマンドを使用します。

() と<条件>の間にはそれぞれ1個以上のスペースが必要です。

```
// コンソールに 1 から 10 まで表示する
var i
1 i @代入
( i @参照 10 <= )
i @参照 .
i @参照 1 add i @代入
loop
```

スクリプトを終了させる

ver. 5.0.0.0 以降

スクリプトの終了は、必ず exit を使用してください。 exit コマンドがないときはエラーとなります。 exit コマンドはパラメータを取りません。

サンプルコード

exit

エラー終了

ver. 5.0.0.0 以降

編集スクリプトをユーザ定義のエラーコードで終了させるには abend を使用します。

コマンド名:abend

パラメータ:終了コード(整数型)

返り値 : なし

サンプルコード

102 abend // エラーコード「102」を表示してこの編集スクリプトを終了させる

※コンソール表示例

ESL:/@C/EDIAN10A/EDIDATA/ediesl/test.esl

ESL:9 行目実行時 ユーザ定義のエラー終了(102)です。

ESL 実行でエラーが発生しました。

論理演算

ver. 5.0.0.0 以降

if、loop 文の条件で使用できる論理演算には以下のものがあります。 真偽の値は整数型の値で表現されます。(0 以外の値:真、0:偽)

●論理積 (AND) &&

2つの整数値が真 (0 以外の値) のとき、スタックのトップに真 (1) が積まれ、そうでないときは、偽 (0) が積まれます。

コマンド名:&&

パラメータ:整数値1(整数型)整数値2(整数型)

返り値 : 真または偽(整数型)

●論理和 (OR) !!

整数値1または整数値2が真(0以外)のとき、スタックのトップに真(0以外)が積まれ、そうでないときは、偽(0)が積まれます。

コマンド名: !!

パラメータ:整数値1(整数型)整数値2(整数型)

返り値 : 真または偽(整数型)

サンプルコード

32>43>&& // スタックのトップに真(1) が積まれる

33 == 23 == # // スタックのトップに真(1) が積まれる

32<43>& // スタックのトップに偽(0) が積まれる

32 == 23 == # // スタックのトップに偽(0) が積まれる

比較演算子

ver. 5.0.0.0 以降

2つのオブジェクトの同値、大小関係を判定するには比較演算子を使用します。

オブジェクトの型は、文字列型同士、単位付き数値文字列同士、または数値(整数型、 実数型)同士の組み合わせを取ることが可能です。

単位付き数値文字列型の場合は、単位が同じである必要があります。

コマンド実行後、真/偽(1/0)がスタックのトップに積まれます。

コマンド名:大小比較 <

大小比較

大小比較 <=

大小比較 >=

等値比較 ==

非等値比較 !=

パラメータ:オブジェクト1 オブジェクト2

(文字列型同士、単位付き数値文字列同士、数値(整数型、実数型)同士)

返り値 : 真または偽(整数型)

サンプルコード

23< // スタックのトップに真(1) が積まれる

23> // スタックのトップに偽(0) が積まれる

23 <= // スタックのトップに真(1) が積まれる

23>= // スタックのトップに偽(0) が積まれる

23 == // スタックのトップに偽(0) が積まれる

23!= // スタックのトップに真(1) が積まれる

// 変数 b よりも変数 a の方が大きかったら、「a の方が大きい」とコンソールに出力する

var a

var b

2a @代入 1b @代入

a @参照 b @参照 > if

"aの方が大きい".

endif

数值演算

ver. 5.0.0.0 以降

数値演算には、加減乗除の4つの演算(add、sub、mul、div)を使用できます。

これらのコマンドは常に2つのパラメータを取り、演算結果はデータスタックに積まれます。

コマンド名:加算 add

減算 sub 乗算 mul 除算 div

パラメータ:オブジェクト1 オブジェクト2

返り値 :整数型、実数型、文字列型のいずれか

パラメータに許される型の組み合わせと、演算結果の型は以下の通りです。

パラメータの型	演算結果
整数型同士	整数型
実数型同士	実数型
整数型と実数型	実数型
単位付き文字列型同士	文字列型(単位付き数値文字列型)

単位付き数値文字列型同士の加減算でサポートしている単位は、「mm」、「pt」、「ees」、「Q」、「H」、乗算では、「mm」、「pt」、「ees」、「Q」、「H」、「%」または単位なしとなります。

除算では、文字列型同士の演算をサポートしていません。

演算結果の単位は、「mm | 固定となります。

サンプルコード

1 2 add // 3 がデータスタックに積まれる

2 1 sub // 1 がデータスタックに積まれる

22 mul // 4 がデータスタックに積まれる

10 2 div // 5 がデータスタックに積まれる

"1.00mm" "1.5pt" add // "1.529167mm"がデータスタックに積まれる

データスタックの操作コマンド

ver. 5.0.0.0 以降

データスタックを操作するには以下のコマンドが使用できます。

処理	コマンド	処理
ドロップ	drop	スタックのトップにあるオブジェクトを捨てる。
複製	dup	スタックのトップにあるオブジェクトの複製を生成しスタックに積む。
ロール	roll	スタックのトップからn個のオブジェクトをm個回転する。
入れ替え	swap	スタックのトップとその次のオブジェクトを入れ替える。
複写	сору	スタックのトップから指定された数の個数のオブジェクトの複製をス
		タックに積む。
		数値に0以下の値を指定した場合はエラーとなる。
位置指定複製	index	スタックのトップから指定した数値番目(0 origin)のオブジェクトの
		複製をスタックに積む。
		数値に0以下の値を指定した場合は、エラーとなる。

●スタックのトップにあるオブジェクトを捨てる

コマンド名: drop

パラメータ:任意のオブジェクト

返り値 : なし

●スタックのトップにあるオブジェクトの複製を生成しスタックに積む

コマンド名:dup パラメータ:なし

返り値 : スタックのトップにあったオブジェクト

●スタックのトップから n 個のオブジェクトを m 個回転する

コマンド名:roll

パラメータ:オブジェクト1 オブジェクト2...オブジェクトn

対象オブジェクト数(整数型) 回転させる回数(整数型)

返り値:n個のオブジェクト(オブジェクト1~オブジェクトn)

●スタックのトップとその次のオブジェクトを入れ替える

コマンド名:swap

パラメータ: オブジェクト1 オブジェクト2 返り値 : オブジェクト2 オブジェクト1

●スタックのトップから指定された数の個数のオブジェクトの複製をスタックに積む

コマンド名: copy

パラメータ:複製の個数(整数型)

返り値 : オブジェクト オブジェクト... (パラメータで指定された数)

●スタックのトップから指定した数値番目のオブジェクトの複製をスタックに積む

コマンド名:index

パラメータ:位置指定(整数型)

返り値 :指定された位置のオブジェクト

サンプルコード

2 3 drop // スタックのトップは2となる (スタック:23 → 2)

23 dup // スタックの内容は23 → 233となる

1 2 3 4 5 6 6 1 roll // スタックの内容は 1 2 3 4 5 6 → 6 1 2 3 4 5 となる

1 2 3 4 5 6 6 -1 roll // スタックの内容は 1 2 3 4 5 6 → 2 3 4 5 6 1 となる

100 200 swap // 以下の実行により、スタックの内容は 100 200 → 200 100 となる

1 2 3 4 5 6 3 copy // スタックの内容は1 2 3 4 5 6 → 1 2 3 4 5 6 4 5 6となる

1 2 3 4 5 6 2 index // スタックの内容は 1 2 3 4 5 6 → 1 2 3 4 5 6 4 となる

配列、辞書を作成する

ver. 5.0.0.0 以降

●配列

配列を作る場合には「[]」を使ってそのオブジェクトを列挙します。

[要素1 要素2・・・要素n]

要素1~n は任意のオブジェクト

●辞書

辞書を作る場合には「<< >>」と「/<key 名>」を使用します

<< /Key名1 要素1 /Key名2 要素2・・・/Key名n 要素n >>

要素1~n は任意のオブジェクト

サンプルコード

[123] // 配列を作成

<</a1/b2/c3>> // 辞書を作成

配列、辞書に要素を追加する

ver. 5.0.0.0 以降

配列、辞書へ要素の追加、変更を行うには put を使用します。

●配列

コマンド名: put

パラメータ:要素を追加する配列(配列型)要素位置(整数型)追加するオブジェクト

返り値 :なし

指定した配列の要素位置(0 origin)に、オブジェクトを追加します。 指定位置にすでにオブジェクトがあった場合、その値は置き換えられます。

●辞書

コマンド名: put

パラメータ:要素を追加する辞書(辞書型) key 名 追加するオブジェクト

返り値 :なし

辞書に指定した key 名でオブジェクトを追加します。 すでに同じ key 名を持つ要素があった場合、その値は置き換えられます。

サンプルコード

[123]24 put // 配列の内容は[124]となる <</pre>

配列、辞書の要素を取得する

ver. 5.0.0.0 以降

配列、辞書から要素のコピーを取得するときには get 使用します。

●配列

コマンド名:get

パラメータ:要素を取得する配列(配列型)要素位置(整数型)

返り値 :オブジェクト

指定した配列の要素位置(0 origin)にあるオブジェクトのコピーがスタックのトップ に積まれます。

●辞書

コマンド名:get

パラメータ:要素を取得するする辞書(辞書型) key 名

返り値 :オブジェクト

辞書から、指定した key 名とペアになっているオブジェクトのコピーがスタックのトップに積まれます。

サンプルコード

["10.0 mm" "20.0 mm" "30.0 mm"] 1 get // スタックのトップに [20.0 mm] が積まれる << /a 1 /b 2 >> /b get // スタックのトップに 2 が積まれる

配列、辞書の要素数を取得する

ver. 5.0.0.0 以降

辞書または配列の要素の数を取得するには length を使用します。

辞書の要素は、名前とオブジェクトの1つのペアを1つの要素として数えます。

コマンド名: length

パラメータ:辞書型または配列型 返り値:要素数(整数型)

サンプルコード

<< /a 1 /b 2 >> length // スタックのトップに2が積まれる

[123] length // スタックのトップに3が積まれる

文字列を連結する

ver. 5.0.0.0 以降

2つの文字列を連結するには strcat を使用します。

コマンド名: strcat

パラメータ:文字列1(文字列型)文字列2(文字列型)

返り値:連結した文字列(文字列型)

サンプルコード

"abc" "edf" strcat // スタックのトップに "abcdef" が積まれる

文字列の長さを調べる

ver. 5.0.0.0 以降

文字列の長さを調べたいときは strlen を使用します 取得できる長さはバイト単位で整数型になります。

コマンド名: strlen

パラメータ:対象文字列(文字列型) 返り値 : 文字列長(整数型)

サンプルコード

"abcde" strlen // スタックのトップに 5 が積まれる "あいうえお" strlen // スタックのトップに 10 が積まれる

文字列を比較する

ver. 5.0.0.0 以降

2つの文字列を比較するには strcmp を使用します。演算子==を用いても比較することが出来ます。

●strcmp コマンド

strcmp では比較対象の文字列 2 が文字列 1 よりも小さいときはマイナスの値が、一致しているときはゼロが、大きいときはプラスの値がスタックのトップに積まれます。

コマンド名: strcmp

パラメータ:比較対象の文字列1(文字列型) 比較対象の文字列2(文字列型)

返り値 : 比較結果(整数型)

サンプルコード

"あいうえお" "あいうえお" strcmp // スタックのトップに 0 が積まれる "あいうえお" "あいうえ" strcmp // スタックのトップに 1 が積まれる "あいうえお" "あいうえおか" strcmp // スタックのトップに 1 が積まれる "あいうえお" "かきくけこ" strcmp // スタックのトップに 1 が積まれる

●演算子==

演算子==では比較対象の文字列1と文字列2が一致する場合には真(1)を、一致しない場合には偽(0)がスタックのトップに積まれます。

比較演算子:==

パラメータ:オブジェクト1 オブジェクト2

返り値 :真または偽(整数型)

サンプルコード

"あいうえお" "あいうえお" == // スタックのトップに真(1) が積まれる "あいうえお" "あいうえ" == // スタックのトップに偽(0) が積まれる "あいうえお" "あいうえおか" == // スタックのトップに偽(0) が積まれる "あいうえお" "かきくけこ" == // スタックのトップに偽(0) が積まれる

文字列を部分的に比較する

ver. 5.0.0.0 以降

strncmp を使えば、2つの文字列の先頭からnバイト目までを比較することができます。 文字列2が文字列1より小さいときはマイナスの値が、一致しているときはゼロが、大 きいときはプラスの値がスタックのトップに積まれます。

コマンド名: strncmp

パラメータ: 比較対象の文字列1(文字列型) 比較対象の文字列2(文字列型)

先頭からのバイト数(整数型)

返り値 : 比較結果(整数型)

```
"abcde" "abc" 3 strncmp // スタックのトップに 0 が積まれる
"abcde" "abc" 4 strncmp // スタックのトップにー1 が積まれる
"abc" "abcde" 4 strncmp // スタックのトップに 1 が積まれる
"abc" "def" 3 strncmp // スタックのトップに 1 が積まれる
"abc" "defg" 4 strncmp // スタックのトップに 1 が積まれる
"abc" "defg" 4 strncmp // スタックのトップに 0 が積まれる
"あいうえお" "あいう" 6 strncmp // スタックのトップに 0 が積まれる
"あいうえお" "あいう" 7 strncmp // スタックのトップに 1 が積まれる
"あいう" "あいうえお" 7 strncmp // スタックのトップに 1 が積まれる
"あいう" "えおか" 6 strncmp // スタックのトップに 1 が積まれる
"あいう" "えおかき" 7 strncmp // スタックのトップに 1 が積まれる
```

文字列を検索する

ver. 5.0.0.0 以降

文字列中に指定した文字列が含まれるかどうか調べたいときは strstr を使用します。 部分文字列が、検索対象文字列のnバイト目 (0 origin) に現れるときにその値がスタックのトップ (整数型) に積まれ、現れないときは、-1 が積まれます。

コマンド名: strstr

パラメータ:検索対象文字列(文字列型) 部分文字列(文字列型)

返り値:部分文字列の出現位置(整数型)

サンプルコード

"abcde" "de" strstr // スタックのトップに3が積まれる

"abcde" "efg" strstr // スタックのトップに-1 が積まれる

"あいうえおかきくけこさしすせそ" "えおかき" strstr // スタックのトップに 6 が積まれる

文字列の一部を取り出す

ver. 5.0.0.0 以降

文字列の一部を、最初の位置と長さを指定して取り出すには substr を使用します。 文字列から、取り出し開始位置 (0 origin) から n バイト目までの部分を取り出しス タックのトップ (文字列型) に積みます。

コマンド名: substr

パラメータ:対象文字列(文字列型)取り出し開始位置(整数型)取り出しバイト数n(整数型)

返り値:開始位置からnバイト目までの文字列(文字列型)

サンプルコード

"abcde" 1 3 substr // スタックのトップに "bcd" が積まれる

数値を文字列に変換する

ver. 5.0.0.0 以降

整数型、実数型のオブジェクトを文字列型に変換する場合には tostring を使用します。

コマンド名: tostring

パラメータ:整数型または実数型 返り値:変換結果(文字列型)

サンプルコード

0.123 tostring // スタックのトップに文字列 "0.123000" が積まれる

1 tostring // スタックのトップに文字列 "1" が積まれる

整数型へ変換する

ver. 5.0.0.0 以降

実数型、文字列型のオブジェクトを整数型のオブジェクトに変換する toint を使用します。

実数型から整数への変換は、「切捨て」で行われます。

単位付き数値文字列を変換した場合には単位は削除されます。

コマンド名: toint

パラメータ: 実数型または文字列型 返り値 : 変換結果(整数型)

サンプルコード

12.5608 toint // スタックのトップに 12 が積まれる

// 単位付き文字列の小数点部分を取り除く

"12.5608mm" toint tostring "mm" streat // スタックのトップに文字列 "12mm" が積まれる

// 単位付き文字列以外の文字列を変換した場合の変換結果

"12.5608あいう" toint // スタックのトップに 12 が積まれる

"12あ" toint // スタックのトップに 12 が積まれる

"a12.5608" toint // スタックのトップに 0 が積まれる

実数型へ変換する

ver. 5.0.0.0 以降

整数型、文字列型のオブジェクトを実数型に変換したい場合は todouble を使用します。 スタックのトップにあるオブジェクトを実数型にして返します。

単位付き数値文字列を変換した場合には単位は削除されます。

コマンド名: todouble

パラメータ:整数型または文字列型 返り値 :変換結果(実数型)

サンプルコード

10 todouble // スタックのトップに実数型で 10.0 が積まれる

"12mm" todouble // スタックのトップに実数型で 12.0 が積まれる

// 単位付き文字列以外の文字列を変換した場合の変換結果

"12.5608あいう" todouble // スタックのトップに実数型で 12.5608が積まれる

"12あ" todouble // スタックのトップに実数型で 12.0 が積まれる

"a12.5608" todouble // スタックのトップに実数型で 0.0 が積まれる

浮動小数点以下を丸める

ver. 5.0.0.0 以降

小数点以下を丸めるには以下のものを使用します。

●小数点以下を切り捨てる

コマンド名: floor

パラメータ: 実数型または文字列型

返り値 : 実数型

●小数点以下を繰り上げる

コマンド名:ceil

パラメータ: 実数型または文字列型

返り値 : 実数型

サンプルコード

5.9 floor // スタックのトップに値が 5.0 の実数型オブジェクトが積まれる5.9 ceil // スタックのトップに値が 6.0 の実数型オブジェクトが積まれる

オブジェクトの型を調べる

ver. 5.0.0.0 以降

オブジェクトの型を取得するには **object** を使用します。 スタックのトップにあるオブジェクトの型を文字列で返します。

コマンド名: object

パラメータ:任意の型のオブジェクト

返り値: "辞書"、"配列"、"文字列"、"整数"、"実数"、"その他"のいずれか(文字列型)

サンプルコード

ファイルを作成する

ver. 5.0.0.0 以降

ファイルを作成するには open を使用します。

オープンの際に、ファイルが存在していないときは新規にファイルが作成され、既に存在しているときはそれまでの内容がクリアされます。

コマンド名:open

パラメータ:ファイル名(文字列型)

返り値 :ファイル型

ファイル名が「/」で始まっていない場合は相対パスとして判断され、環境変数 EDI_ESL_FILEPATH からの相対パスとなります。環境変数 EDI_ESL_FILEPATH が未設 定の場合は環境変数 EDITEXT からの相対パスとなります。

ファイル名が「./|で始まっている場合も環境変数からの相対パスになります。

サンプルコード

"/@D/data/書き出し.txt" open // 絶対パスで「D:\u00e4data\u00e4書き出し.txt」をオープン

"書き出し.txt" open // 相対パスで「書き出し.txt」をオープン

ファイルを閉じる

ver. 5.0.0.0 以降

open で開いたファイルを閉じるには close を使用します。

コマンド名: close

パラメータ:ファイル型

返り値 :なし

サンプルコード

var fileObj

"sample.txt" open fileObj @代入

fileObj @参照 close

ファイルへ文字列を書き込む

ver. 5.0.0.0 以降

文字列をファイルへの書き込むには putstring を使用します。

コマンド名: putstring

パラメータ:文字列型 ファイル型

返り値 : なし

サンプルコード

// 選択した枠があふれていたら c:\take data\take afure.log に "あふれました。" を書き出す

"/@C/data/afure.log" open dup

@あふれ評価 if "あふれました。" swap putstring endif

close

改行コードを使う

ver. 5.0.0.0 以降

編集スクリプトではファイルに書き込む際に、2種類の改行コードを使用することができます。

NL

NL は改行コード (0xa) を生成して、文字列型オブジェクトとしてスタックのトップ に積みます。

コマンド名:NL パラメータ:なし

返り値 : 改行コード(文字列型)

● CRLF

CRLF は改行コード(0x0d0a)を生成して、文字列型オブジェクトとしてスタックのトップに積みます。

コマンド名: CRLF パラメータ: なし

返り値 : 改行コード(文字列型)

サンプルコード

// ファイルに改行コードを書き込む

var fileObj

"sample.txt" open fileObj @代入

NL fileObj @参照 putstring

CRLF fileObj @参照 putstring

fileObj @参照 close

ファイルが存在しているかチェックする

ver. 5.0.0.0 以降

ファイルが存在してるか調べるには access を使用します。

与えられたファイル名のファイルが存在するかどうかを確認するためのコマンドで、ファイルが存在するときはスタックのトップに真(1)が、存在しないときは偽(0)が積まれます。

コマンド名: access

パラメータ:ファイル名(文字列型) 返り値:真または偽(整数型)

サンプルコード

// c:¥data¥test.txt が存在していたらファイルを開く

"/@C/data/test.txt" dup access if open

close

endif

検証者 memo

ファイル名が「/」で始まっていない場合に環境変数による相対パスが適用されるのかについては未検証。 「./| の場合も同様。

枠を選択する

ver. 5.0.0.0 以降

指定した名称の枠を選択するには@枠選択を使用します。

枠選択の対象は固定枠、表枠、小組枠、文字領域、画像領域、表領域です。

コマンド名:@枠選択

パラメータ: 枠名称(文字列型)

返り値 : なし

●エラー制御

指定された名称の枠が無い等で**[®]枠選択**実行時にエラーが発生した場合、**[®]枠選択条件**で 設定した条件に従って処理されます。

@枠選択条件が設定されていない場合にはエラー発生時に処理が終了されます。

エラー時継続時に**0枠選択**が失敗した場合は、**0枠選択**実行前の選択状態は変更されません。

コマンド名:@枠選択条件

パラメータ: "エラー時終了" または "エラー時継続"(文字列 型)

エラー時終了: エラー時、編集スクリプトを終了する エラー時継続: エラー時、編集スクリプトを継続する

返り値 :なし

サンプルコード

"エラー時継続" @枠選択条件 // エラーが出ても処理を終了しない

"A000" @枠選択 // 枠名称 "A000" の枠を選択する

枠の選択を解除する

ver. 5.0.0.0 以降

枠の選択を明示的に解除するには**@枠解除**を使用します。

選択対象を本文にするときに使用します。

コマンド名:@枠解除

パラメータ: なし 返り値 : なし

サンプルコード

@枠解除

枠選択状態を取得する

ver. 5.0.0.0 以降

編集スクリプト実行時、枠が選択されているかを調べるには**0枠選択状態**を使用します。 枠が選択状態のときは真(1)が、そうでないときは偽(0)が、スタックのトップに積 まれます。

コマンド名:@枠選択状態

パラメータ: なし

返り値 :真または偽(整数型)

サンプルコード

```
// 枠が選択されていたら、コンソールに "選択中" を出力
@枠選択状態 if "選択中" . endif
// 枠を A000 があるか確認してメッセージを表示する
"エラー時継続" @枠選択条件
                      // エラーが出ても処理を継続する
@枠解除
                      // すでに選択されている枠があったら解放
"A000" @枠選択
                      // 枠名称 "A000" の枠を選択する
@枠選択状態 if
                      // 枠が選択されていたら
 "A000 を選択しました".
else
                      // 枠が選択されていなかったら
 "A000 は見つかりませんでした".
endif
```

連結した小組枠を選択する

ver. 5.0.0.0 以降

0枠選択コマンドで選択された枠が連結した小組枠であった場合、最初の小組枠のみが 選択された状態になります。そのため、2番目以降の小組枠を選択したい場合には**0小組枠 枠選択**を使用する必要があります。

あふれているところは選択の対象になりません。

コマンド名:@小組枠選択

パラメータ: "次へ"、"前へ"、"最初へ"、"最後へ" のいずれか(文字列型)

返り値 : なし

最初の枠を選択時の"前へ"、最後の枠を選択時の"次へ"は無効となります。

サンプルコード

// 現在選択している連結小組枠の最後の小組枠を選択する。

"最後へ"@小組枠選択

枠のサイズを取得する

ver. 5.0.0.0 以降

選択されている枠の幅、高さを取得するには**0枠サイズ**を使用します。 矩形サイズ辞書の形式でスタックのトップに積まれます。

コマンド名:@枠サイズ

パラメータ:なし

返り値 : 矩形サイズ辞書(辞書型)

矩形サイズ辞書

属性名	属性値の型	説明
/幅	文字列	矩形の幅(単位 ees の単位付き数値文字列)
/高さ	文字列	矩形の高さ(単位 ees の単位付き数値文字列)

サンプルコード

// 矩形サイズを取得し幅と高さをコンソールに表示する

@枠サイズ dup // 矩形サイズ辞書がスタックのトップに積まれる

/幅 get . // コンソールに幅を表示 /高さ get . // コンソールに高さを表示

枠の位置を取得する

ver. 5.0.0.0 以降

選択されている枠の位置情報を取得するには**0枠位置**を使用します。 座標値辞書の形式で、スタックのトップに積まれます。

コマンド名:@枠位置

パラメータ:なし

返り値 :座標値辞書(辞書型)

座標値辞書

属性名	属性値の型	説明
/X座標	文字列	X 方向の座標値(単位 ees の単位付き数値文字列)
/Y座標	文字列	Y 方向の座標値(単位 ees の単位付き数値文字列)

サンプルコード

// 位置情報を取得し X 座標と Y 座標をコンソールに表示する

@枠位置 dup // 矩形サイズ辞書がスタックのトップに積まれる

 /X座標 get .
 // コンソールに X 方向の座標値を表示

 /Y座標 get .
 // コンソールに Y 方向の座標値さを表示

枠の属性を変更する

ver. 5.0.0.0 以降

選択されている枠の枠名称、座標値、サイズを変更するには**@枠属性変更**を使用します。 但し、表枠は枠名称のみが変更可能です。

コマンド名:@枠属性変更

パラメータ: 枠属性辞書(辞書型)

返り値 :なし

枠属性辞書

属性名	属性値の型	説明	備考
/座標値	辞書	領域の左上の座標値を指定する(座標値辞書)	省略時、そのまま
/矩形サイズ	辞書	枠、領域の幅、高さを指定する(矩形サイズ辞書)	省略時、そのまま
/回転角	文字列	文字列領域の回転角を指定する	省略時、そのまま 回転出来ないものに指定 されたときは、無視する
/枠名称	文字列	枠、領域の枠名称を変更する	省略時、そのまま
/レイヤ番号	文字列/整数	枠、領域のレイヤ番号を変更する 整数は 1~8 文字列は、"1"、"2""8"	省略時、そのまま
/行間	配列	小組枠、文字領域の行間を指定する [行間種別 行間値] 行間種別: "アキ"、"TOP"、 "CENTER"、"BOTTOM" のいずれか 行間値: 長さを表す単位付き数値文字列	省略時、そのまま
/変形	配列	小組枠、文字領域の変形率を指定する [変形種別 変形率] 変形種別: "平体"、"長体" のいずれか 変形率: %(ASCII)で終わる文字列	省略時、そのまま
/共通体裁	辞書	文字サイズ、書体等を指定する(共通体裁辞書)	省略可

座標値辞書

属性名	属性値の型	説明	備考
/X座標	文字列	X 方向の座標値(単位は mm、pt、ees)	省略可不可
/Y座標	文字列	Y 方向の座標値(単位は mm、pt、ees)	省略可不可

矩形サイズ辞書

属性名	属性値の型	説明	備考
/幅	文字列	矩形の幅(単位は mm、pt、ees)	省略不可
/高さ	文字列	矩形の高さ(単位は mm、pt、ees)	省略不可

共通体裁辞書

属性名	属性値の型	説明	備考
/和文書体	文字列/整数	和文の書体を指定する*1)	省略可
/欧文書体	文字列/整数	欧文の書体を指定する*1)	省略可
/和文サイズ	文字列	和文の文字サイズを指定する	省略可
		(単位は mm、pt、H、Q)	
/欧文サイズ	文字列	欧文の文字サイズを指定する	省略可
		(単位は mm、pt、H、Q)	
/斜体	文字列	斜体を指定する "xx 度" または "なし"	省略可
		xx は-30~30 までの 5 刻みの整数	
/変形	文字列	変形を指定する "長体 xx%"、"平体 xx%"、"なし"	省略可
		xx は 0~設定された変形上限までの整数	
/文字間	文字列	文字間を指定する*2)	省略可
/行間	文字列	行間を指定する*2)	省略可
/かな詰め	文字列	かな詰めを指定する	省略可
		"あり"、"する"、"なし"、"しない"、の何れか	
/行揃え	文字列	行揃えを指定する	省略可
		"上"、"中"、"下"、"行どり"(行取りも可) の何れか	
/字揃え	文字列	字揃えを指定する	省略可
		"左"、"中"、"右"、"字どり"(字取りも可) の何れか	

- *1) 文字列で書体名か整数値で書体番号のいずれかを指定する。
 - 例) 文字列: "<C> A-OTF 新ゴ Pro L" 整数: 9036
- *2) 値は、"<長さ>:<アキ/送り指定>"の形式を取る。

<長さ>: 単位付き数値文字列(単位は、mm,pt,H,Q)。

<アキ/送り指定>: "アキ"、"TOP"、"CENTER"、"BOTTOM" のいずれかで省略値は "アキ" となる。 文字間では、"BOTTOM" を指定するとエラーとなる。

例) "1.00pt:アキ"

サンプルコード

<< /枠名称 "sample" >> @枠属性変更

選択枠を「部品書出」する

ver. 5.0.0.0 以降

選択されている枠に編集機能の「部品書出」を適用するには@部品書出を使用します。

コマンド名:@部品書出

パラメータ:部品書出辞書(辞書型)

返り値 :なし

部品書出辞書

属性名	属性値の型	説明	備考
/ファイル名	文字列	部品のファイル名を指定する。拡張子は不要。	必須

ファイル名には相対パスまたは絶対パスが使用できます。

相対パスで指定した場合には環境変数 EDIPARTS からの相対パスになります。 どちらの場合も拡張子は不要です。

サンプルコード

// ファイル名 T0020.PAR に部品書出を行う

<< /ファイル名 "T0020" >> @部品書出 // 相対パスで指定

<< /ファイル名 "/@C/T0020" >> @部品書出 // 絶対パスで指定

選択枠を EPS 画像にする

ver. 5.0.0.0 以降

選択されている枠を EPS 画像として書き出すには**@EPS変換**を使用します。

コマンド名:@EPS変換 ※EPSの記述は全角です

パラメータ: EPS変換辞書(辞書型)

返り値 : なし

EPS変換辞書

属性名	属性値の型	説明	備考
/ファイル名	文字列	EPS ファイル名を指定する。拡張子は不要。	必須
/オプション	文字列	EPS 変換オプション定義ファイルのファイル	省略時は設定
		名を指定する。	の標準値

EPSファイル名およびオプション定義ファイル名は、それぞれ絶対パスか相対パスで 指定できます。

EPS ファイル名を相対パスで指定した場合には環境変数 EDIPSDIR からの相対パスになり、オプション定義ファイル名を相対パスで指定した場合には環境変数 EDI_ESL_TEXT からの相対パスになります。

サンプルコード

// EPS 変換オプション定義ファイル OPT0020.txt を使って EPS 画像を作成する

<<

/ファイル名 "T0020"

/オプション "OPT0020.txt"

>> @EPS 変換

●EPS 変換オプション定義ファイル

オプションファイルは以下の仕様に基づきテキストファイルを作成します。

<キーワード>=<値><改行>という1行で1つの定義を行います(=は半角のみ)。

定義が省略された場合は、立ち上げ直後の EPS 変換パネルの値となります。

半角空白文字やタブ文字は無視されます。

半角の#から<改行>まではコメントとなります。

キーワードとその値は以下の通りです。

サムネール= {なし | 低解像度 | 中解像度 | 高解像度 | 低 | 中 | 高}

変換対象= {ページ全体 | 固定枠 | ページ | 枠}

裁ち落し=<裁ち落し幅>

※単位なしの数値は2400dpi。単位にmm, cm, inch, q, pt, neu が指定可 (例:1.5cm)。

あふれページ= {はい | いいえ}

空白ページ= {はい | いいえ}

PostScript = $\{1 \mid 2 \mid JPEG\}$

※1: PostScript レベル2がオフ

※2: PostScript レベル2がオン、JPEG フォーマットがオフ

※JPEG: PostScript レベル2がオン、JPEG フォーマットがオン

欧文フォント埋め込み= {はい | いいえ}

※パネルでは、「欧文書体 PS フォントの埋め込みを許す」

フチ・影・文字色なしを白= {はい | いいえ}

※パネルでは、「フチ/影付きで文字色なしを「白」で描画」

別版出力=<別版指定文字列>

※<別版指定文字列>に何も指定しなければ、別版出力「しない」

倍率=<倍率>

外字の出力形式= {標準 | 全て Type 3 フォント | ユーザ外字、面種のみ Type 3 フォント | ユーザ外字のみ Type 3 フォント | 面種のみ Type 3 フォント | 全て Type 1 フォント |

警告エラー= {表示のみ | 変換後に確認 | 変換を中止}

TRUETYPE の出力形式= {外字 | Type42形式}

※ @EPS変換では「変換対象」「裁ち落し」の設定があるとエラーになる

サンプルコード

サムネール="低解像度"

あふれページ="いいえ"

空白ページ="いいえ"

PostScript = 2

欧文フォント埋め込み="はい"

フチ・影・文字色なしを白="いいえ"

別版出力=

倍率=100

外字の出力形式="全て Type3 フォント"

警告エラー="変換後に確認"

TRUETYPE の出力形式="Type42 形式"

選択枠のプレビュー画像を作成する

ver. 5.0.0.0 以降

選択されている枠をプレビュー画像として書き出すには**0プレビュー出力**を使用します。 対象となる枠は固定枠、小組枠、表領域で、枠の領域サイズに切り取られた画像がプレ ビュー画像(jpg 形式)として書き出されます。

コマンド名:@プレビュー出力

パラメータ:プレビュー出力辞書(辞書型)

返り値 : なし

プレビュー出力辞書

属性名	属性値の型	説明	備考
/ファイル名	文字列	プレビュー画像のファイル名を指定。拡張子は不要。	必須
/オプション	文字列	プレビュー画像作成オプション定義ファイルのファイル名	省略可
		を指定する。	

画像ファイル名およびオプション定義ファイル名は、それぞれ絶対パスか相対パスで指定できます。

画像ファイル名ファイル名を相対パスで指定した場合には Edian インストールフォル ダからの相対パスになり、オプション定義ファイル名を相対パスで指定した場合には環境 変数 EDI ESL TEXT からの相対パスになります。

サンプルコード

// オプション定義ファイル OPT0020.txt を使ってファイル名 T0020.JPG にプレビュー画像を出力する

<<

/ファイル名 "T0020"

/オプション "OPT0020.txt"

>> @プレビュー出力

●プレビュー画像作成オプション定義ファイル

オプションファイルは以下の仕様に基づきテキストファイルを作成します。

<キーワード>=<値><改行>という1行で1つの定義を行います(=は半角のみ)。

値が文字列の場合は"で囲みます

キーワードとその値は以下の通りです。

半角の#から<改行>まではコメントとなります。

DPI=<作成する画像の DPI>

※7~2400、省略値は100

画像種別= {JPEG | GIF | TIFF | TIFF/256 | BMP | BMP/256}

※作成する画像の画像種別、省略値は JPEG。

「TIFF/256」は、TIFF の 256 色パレットカラー、「BMP/256」は BMP の 256 色パレットカラー。

出力レイヤ=<出力レイヤ指定>

※出力対象のレイヤ、数値と一(範囲), (列挙)で指定する。

(例:1-2,4-5)、省略値は1-8

裁ち落し=<裁ち落し幅>

※画像を周囲に付加する余白を mm、pt の単位付き数値で指定する。

単位無しは 1/2400 インチ単位、省略値は 0

※裁ち落し設定はマニュアルに記載されているが、枠選択からのプレビューからは裁ち落しが設定できないため、記述されているとエラーになる

サンプルコード

DPI=300

画像種別="JPEG"

出力レイヤ="1-8"

固定枠を作成する

ver. 5.0.0.0 以降

固定枠を作成するには@固定枠作成を使用します。

作成されるページは現在開かれているページ(カレントページ)になります。

コマンド名:@固定枠作成

パラメータ:固定枠辞書(辞書型)

返り値 :なし

固定枠辞書

属性名	属性値の型	説明	備考
/座標值	辞書	固定枠の左上の座標値を指定する(座標値辞書)	必須
/枠名称	文字列	固定枠の枠名称を指定する。	必須
/矩形サイズ	辞書	固定枠の幅、高さを指定する(矩形サイズ辞書)	必須

座標值辞書

属性名	属性値の型	説明	備考
/X座標	文字列	X 方向の座標値(単位は mm、pt、ees)	必須
/Y座標	文字列	Y 方向の座標値(単位は mm、pt、ees)	必須

矩形サイズ辞書

属性名	属性値の型	説明	備考
/幅	文字列	矩形の幅(単位は mm、pt、ees)	必須
/高さ	文字列	矩形の高さ(単位は mm、pt、ees)	必須

サンプルコード

// 左から 10.0mm、天から 20.0mm の位置に幅 190.0mm、高さ 250.0mm の固定枠を作成する

<<

/座標値 << /X座標 "10.0mm" /Y座標 "20.0mm" >>

/枠名称 "koteiwaku"

/矩形サイズ << /幅 "190.0mm" /高さ "250.0mm" >>

>> @固定枠作成

枠種別の取得する

ver. 5.0.0.0 以降

選択された枠の枠種別を取得するには**0枠種別取得**を使用します。 枠種別は、文字列でスタックのトップに積まれます。

コマンド名:@枠種別取得

パラメータ:なし

返り値 : "小組枠"、"文字領域"、"画像"、"固定枠"、"表"、"連結表"、"浮動表"、"その他"

のいずれか(文字列型)

枠名称の一覧を取得する

ver. 5.0.0.0 以降

記事中にある枠名称の一覧を取得するには**0枠名称一覧**を使用します。 枠名称の一覧は、配列型でスタックのトップに積まれます。

コマンド名:@枠名称一覧

パラメータ:なし

返り値 : 枠名称の一覧(配列型)

サンプルコード

// 記事中の枠名称一覧から特定フォルダの画像と枠種別が画像の枠を入れ替える var a var n var num var dir var gazou // 入替え用画像の置き場所を設定する。 "EDIGAZOU" @環境変数 // 標準画像フォルダを環境変数から取得する "/esl 画像/" strcat dir @代入 @枠名称一覧 dup a @代入 // 記事中の枠名称の一覧を変数 a に代入する a @参照 length num @代入 // 選択した枠の数を取得 0 n @代入 // n は、0 から num-1 までループ (n @参照 num @参照 <) a @参照 n @参照 get dup // 配列からn番目の元画像名を取得 // "dir+元画像名" が実際に存在していたら画像入替を行う dir @参照 swap streat dup gazou @代入 access if dup @枠選択 strlen 0 > @枠種別取得 "画像" == & & if <<

```
/ファイル名 gazou @参照
/入替え位置 "画像の中心"
/入替えサイズ "枠"
>> @画像入替
endif
endif
n @参照 1 add n @代入
loop
```

枠の余白を取得する

ver. 5.0.0.0 以降


選択された枠の余白を調べるには@枠余白を使用します。

余白の値は上下左右の順に並べられた配列型でスタックのトップに積まれます。 表、小組枠、文字領域で使用できます。

コマンド名:@枠余白 パラメータ:なし

返り値 : 上下左右の余白値(配列型) ※単位は ees

枠



選択中の枠から次の枠を選択する

ver. 5.0.0.0 以降

@次枠選択を使用すると選択中の枠の次の枠から、指定した枠名称の枠を検索し選択することができます。

検索順序は機能「内容一覧」と同じになります。

枠が選択されていないときは、記事の先頭から検索されます(@枠選択と同じ結果)。

指定された枠名称を持つ枠が見つかったときは真(1)が、そうでないときは、偽(0)が、スタックのトップに積まれます。

コマンド名:@次枠選択

パラメータ: 枠名称(文字列型) 返り値 : 真または偽(整数型)

サンプルコード

// 記事中の枠名称 "A" を持つ全ての枠の枠種別を取得し、コンソール・ログに出力する

@枠解除 // 枠の選択状態の影響を回避するため、選択を解除しておく

("A"@次枠選択)

@枠種別取得.

loop

カットバッファを変更する

ver. 5.0.0.0 以降

カットバッファを変更するには**@カットバッファ選択**を使用します。

コマンド名:@カットバッファ選択

パラメータ:カットバッファ番号(整数型) ※1~10まで

返り値 : なし

サンプルコード

10 @カットバッファ選択 // カットバッファを 10 に変更する

選択中の枠を削除する

ver. 5.0.0.0 以降

選択中の枠を削除するには@枠削除を使用します。

コマンド名:@枠削除 パラメータ:なし 返り値:なし

選択中の枠をカットする

ver. 5.0.0.0 以降

選択中の枠をカットするには@枠切取りを使用します。

コマンド名:@枠切取り

パラメータ:なし

返り値 :なし

コピーした枠をペーストする

ver. 5.0.0.0 以降

コピーまたはカットした枠をペーストするには**@枠貼込み**を使用します。 ペーストされるページは現在開かれているページ (カレントページ) になります。 カレントバッファが空の場合はエラーとなり、処理を終了します。

コマンド名:@枠貼込み

パラメータ:座標値辞書(辞書型)

返り値 : なし

座標値辞書

属性名	属性値の型	説明	備考
/X座標	文字列	X 方向の座標値(単位は mm、pt、ees)	必須
/Y座標	文字列	Y 方向の座標値(単位は mm、pt、ees)	必須

サンプルコード

// 用紙の左上から(10mm,20mm)の位置に、カレントバッファの内容を張り込む << /X 座標 "10mm" /Y 座標 "20mm" >> @枠貼込み

オブジェクトを全選択する

ver. 5.0.0.0 以降

カレントページにある全てのオブジェクトを選択するには**@オブジェクト全選択**を使用 します。

選択されるものは編集画面の「編集」→「全選択」を実行したときと同じになります。

コマンド名:@オブジェクト全選択

パラメータ:なし

返り値 :選択したオブジェクトの数(整数型)

サンプルコード

// オブジェクトを全選択しファイル名 PARTS.PAR で部品に書き出す

@オブジェクト全選択

<<

/ファイル名 "PARTS"

>> @部品書出

基準点を設定する

ver. 5.0.0.0 以降

基準点設定の仕方には2通りの方法がありますが、座標値から基準点を設定する場合も、 選択中の枠から基準点を設定する場合も**@基準点設定**を使用します。

●座標値から基準点を設定

コマンド名:@基準点設定

パラメータ:座標値辞書(辞書型)

返り値 : なし

座標値辞書

属性名	属性値の型	説明	備考
/X座標	文字列	X 方向の座標値(単位は mm、pt、ees)	必須
/Y座標	文字列	Y 方向の座標値(単位は mm、pt、ees)	必須

●選択中の枠から基準点を設定

コマンド名:@基準点設定

パラメータ:選択基準点辞書(辞書型)

返り値 : なし

選択基準点辞書

属性名	属性値の型	説明	備考
/水平方向	文字列	"左"、"中"、"右"、"なし" のいずれか	必須
/垂直方向	文字列	"上"、"中"、"下"、"なし" のいずれか	必須

基準点に揃える

ver. 5.0.0.0 以降

選択中の枠を基準点に揃えるには@基準点揃えを使用します。

コマンド名:@基準点揃え

パラメータ:選択基準点辞書(辞書型)

返り値 : なし

選択基準点辞書

属性名	属性値の型	説明	備考
/水平方向	文字列	"左"、"中"、"右"、"なし" のいずれか	必須
/垂直方向	文字列	"上"、"中"、"下"、"なし" のいずれか	必須

基準点の位置を調べる

ver. 5.0.0.0 以降

現在の基準点を調べるには@基準点位置を使用します。

コマンド名:@基準点位置

パラメータ:なし

返り値 :座標値辞書(辞書型)

座標値辞書

属性名	属性値の型	説明
/X座標	文字列	X 方向の座標値(単位は ees)
/Y座標	文字列	Y 方向の座標値(単位は ees)

サンプルコード

// 文字列が「+」のセルを順次選択し、セルの左上の座標をコンソールに表示する

"HYO" @枠選択

("+"@次セル選択)

<< /水平方向 "左" /垂直方向 "上" >> @基準点設定

@基準点位置 .

loop

ページを変更する

ver. 5.0.0.0 以降

現在開いているページを変更するには@ページ移動を使用します。

コマンド名:@ページ移動

パラメータ:ページ番号(整数型)

返り値 :なし

ページ番号は、記事属性の「開始ページ」の値から始まる番号のことを指します。 存在しないページ番号を指定した場合、次のような動作をします。

0 またはマイナスの整数を指定した場合	開始ページに移動する
開始ページ未満の場合	開始ページに移動する
10000 以上の整数を指定した場合	終了ページに移動する
終了ページを超えた場合	終了ページに移動する

サンプルコード

// 3ページに移動する。

3 @ページ移動

// 開始ページが奇数か偶数かを調べる。

var page_num

0 @ページ移動

@カレントページ page_num @代入

page_num @参照 dup 0.5 mul toint 2 mul == if

"偶数ページ" . // コンソールに "偶数ページ" と表示

else

"奇数ページ" . // コンソールに "奇数ページ" と表示

endif

現在ページを調べる

ver. 5.0.0.0 以降

現在開いているページを調べるには@カレントページを使用します。

コマンド名:@カレントページ

パラメータ:なし

返り値 :ページ番号(整数型)

ページ番号は、記事属性の「開始ページ」の値から始まる番号のことを指します。

サンプルコード

// 次のページへ移動する。

@カレントページ 1 add @ページ移動

画像を張り込む

ver. 5.0.0.0 以降

画像を張り込みたい場合には@画像取込を使用します。

張り込まれるページは現在開かれているページ(カレントページ)になります。

コマンド名:@画像取込

パラメータ:画像取込み辞書(辞書型)

返り値 : なし

画像取込み辞書

属性名	属性値の型	説明	備考
/座標值	辞書	画像領域の左上の座標値を指定する(座標値辞書)	必須
/ファイル名	文字列	画像のファイル名を指定する	必須
/倍率	文字列	取り込む際の倍率を指定する "1%"~"1000%"	省略時 "100%"
/保持方法	文字列	リンク形式にするかどうかを指定する	省略時
		"リンク形式" または "記事にコピー"	"記事にコピー"
/回転角	文字列	領域の回転角を指定する "0.0"~"360.0"	省略時 "0"

座標値辞書

属性名	属性値の型	説明	備考
/X座標	文字列	X 方向の座標値(単位は mm、pt、ees)	必須
/Y座標	文字列	Y 方向の座標値(単位は mm、pt、ees)	必須

画像のファイル名は、絶対パスか相対パスで指定できます。

相対パスで指定した場合には環境変数 EDIGAZOU からの相対パスになります。

サンプルコード

// g000.eps を用紙の左上から(100mm,200mm)の位置に90度回転して張り込む

<< /座標値 << /X座標 "100mm" / Y座標 "200mm" >>

/ファイル名 "g000.eps"

/回転角 "90"

>> @画像取込

画像領域を作成する

ver. 5.0.0.0 以降

画像領域を作成するには@画像領域作成を使用します。

このコマンドでは画像ファイルを取り込むことはできません。

コマンド名:@画像領域作成

パラメータ:画像領域辞書(辞書型)

返り値 : なし

画像領域辞書

属性名	属性値の型	説明	備考
/座標値	辞書	画像領域の左上の座標値を指定する(座標値辞書)	必須
/元ファイル名	文字列	画像領域の枠名称を指定する	必須
/矩形サイズ	辞書	画像領域の幅、高さを指定する(矩形サイズ辞書)	必須
/回転角	文字列	領域の回転角を指定する "0.0"~"360.0"	省略時 "0"

座標値辞書

属性名	属性値の型	説明	備考
/X座標	文字列	X 方向の座標値(単位は mm、pt、ees)	必須
/Y座標	文字列	Y 方向の座標値(単位は mm、pt、ees)	必須

矩形サイズ辞書

属性名	属性値の型	説明	備考
/幅	文字列	矩形の幅(単位は mm、pt、ees)	必須
/高さ	文字列	矩形の高さ(単位は mm、pt、ees)	必須

サンプルコード

<<

/座標値 << /X座標 "100mm" /Y座標 "200mm" >>

/元ファイル名 "G000"

/矩形サイズ << /幅"20pt" /高さ "30pt" >>

>> @画像領域作成

画像を入れ替える

ver. 5.0.0.0 以降

選択中の画像領域の画像を入れ替えるには@画像入替を使用します。

コマンド名:@画像入替

パラメータ:画像入替辞書(辞書型)

返り値 : なし

画像入替辞書

属性名	属性値の型	説明	備考
/ファイル名	文字列	画像のファイル名を指定する	必須
/入替え位置	文字列	入替え時の位置を指定する。 "左上"、"画像の中心"、"切抜き図形" のいずれか	省略時 "左上"
/保持方法	文字列	リンク形式にするかどうかを指定 "リンク形式" または "記事にコピー"	省略時 "記事にコピー"
/入替えサイズ	文字列	入替え時のサイズを指定する。 "枠"、"枠(縦横等倍)"、"画像"、"枠の倍率" のいずれか	省略時 "枠"

画像のファイル名は、絶対パスか相対パスで指定できます。

相対パスで指定した場合には環境変数 EDIGAZOU からの相対パスになります。

サンプルコード

// 選択した画像領域に対して、ファイル名 g002.eps の画像とリンク形式で入れ替える

<<

/ファイル名 "g002.eps"

/保持方法 "リンク形式"

>> @画像入替

画像をファイルとして取り出す

ver. 5.0.0.0 以降

選択中の画像領域の画像をファイルとして取り出すには@画像取出を使用します。

コマンド名:@画像取出

パラメータ:取り出し先のファイル名(文字列型)

返り値 :なし

取り出し先のファイル名は、絶対パスか相対パスで指定できます。

相対パスで指定した場合にはユーザープロファイルフォルダからの相対パスになります。

例) c:\Document and Settings\wing5

選択している枠のあふれ状態を調べる

ver. 5.0.0.0 以降

選択中の枠のあふれ状態には、選択中の枠自身があふれている場合と選択中の枠の領域内にあふれている枠がある場合の2種類があります。

枠の領域内にあふれている枠があるか調べたい場合には**@あふれ評価**を使用し、枠自身があふれているか調べたい場合には**@枠あふれ評価**を使用します。

どちらも、あふれ状態のときは、真(1)が、そうでないときは、偽(0)が、スタックのトップに積まれます。

コマンド名:@あふれ評価

@枠あふれ評価


パラメータ:なし

返り値 : あふれ状態(あふれていたら真(1)、いなければ偽(0))(整数型)

枠のあふれ状態と返り値

枠のあふれ状態	@あふれ評価	@枠あふれ評価
選択枠自身があふれていて 選択枠領域内にもあふれている枠がある	真(1)	真(1)
選択枠自身はあふれているが 選択枠領域内にはあふれている枠がない	真(1)	真(1)
選択枠自身はあふれていないが 選択枠領域内にはあふれている枠がある	真(1)	偽 (0)
選択枠自身はあふれておらず 選択枠領域内にはあふれている枠がない	偽 (0)	偽 (0)

下図の例で枠Aを選択した場合、枠B、CはAの領域内にある枠と判断されます。



記事全体のあふれ状態を調べる

ver. 5.0.0.0 以降

記事全体のあふれ状態を調べるには0記事全選択を使用したあと、0あふれ評価を使用します。

コマンド名:@記事全選択

パラメータ: なし 返り値 : なし

サンプルコード

@記事全選択

@あふれ評価

あふれ情報をファイルに書き出す

ver. 5.0.0.0 以降

あふれ情報を調べてファイルに書き出すには**@あふれ情報出力**または**@枠あふれ情報出力** を使用します。

選択中の枠の領域内のあふれ情報を書き出したい場合には**@あふれ情報出力**を使用し、 枠自身のあふれ情報書き出したい場合には**@枠あふれ情報出力**を使用します。

記事全体のあふれ情報を書き出したい場合には**@記事全選択**を使用したあと、**@あふれ情報出力**を使用します。何も選択されていない状態ではエラーとなります。

あふれ情報は XML 形式のデータとして書き出されます。

コマンド名:@あふれ情報出力

@枠あふれ情報出力

パラメータ:ファイル名(文字列型)

返り値 : なし

ファイル名は、絶対パスか相対パスで指定できます。

相対パスで指定した場合には環境変数 EDITEXT からの相対パスになります。

サンプルコード

// 枠名称 A000 の枠を選択しその枠のテキストがあふれていたら、あふれ情報をファイル afure.txt に書き出す

"A000" @枠選択

@あふれ評価

if

"afure.txt" @あふれ情報出力

endif

選択している枠の残り行数を取得する

ver. 5.0.0.0 以降

選択中の枠の残り行数を調べたい場合には**@残り行数**を使用します。 枠を選択していない状態でコマンドを実行すると本文が対象になります。

コマンド名:@残り行数

パラメータ:なし

返り値 :残り行数辞書(辞書型)

残り行数辞書

属性名	属性値の型	説明
/1	配列	枠が連結している場合には連結した順に/1・・・が辞書に記述される
/2	配列	先頭の要素は枠の段数を表す
/3	配列	│ │ │ │ │ │ │ 2 番目以降の要素は各段の残り行数を表す │ │ │ │ │ │ │
:	配列	/ /1 [500000] /2 [500555]

サンプルコード

"WAKU" @枠選択 // WAKU という名前の枠を選択して残り行数を取得 @残り行数

// 連結した3つ目の枠の残り行数を取得する

/3 get // 3つ目の枠

1 get // 1 段目の枠の残り行数

// 3行以上残っていたら、

3 >= if

// 残り行数を埋めるテキスト読込

<< /ファイル名 "埋め草.txt" /種別 "トリガー" >> @トリガー自動読込

endif

exit


空領域を取得する

ver. 5.0.0.0 以降

0残りエリアを使用すると枠または本文で文字を流し込むことが可能な領域について、文字が流れていない部分を矩形の領域として取得することができます。

選択した枠または本文中の文字が流れていない矩形 領域と、文字が途中までしか流れていないところの残 りの矩形部分が、枠毎(ページ毎)、段毎に出力されま す。

出力される矩形情報は、見た目の矩形の左上の座標値と幅、高さとなります(組方向に依存しない)。



コマンド名:@残りエリア

パラメータ:なし

返り値 :残りエリア情報辞書(辞書型)

残りエリア情報辞書

属性名	属性値の型	説明
/1	辞書	最初の枠(ページ)の残りエリア用辞書
/2	辞書	2番目の枠(ページ)の残りエリア用辞書
/3	辞書	3番目の枠(ページ)の残りエリア用辞書
i		/1、/2 は半角(ASCII)の数字文字列で、小組枠のときは連結した何番目の小組枠かを表し、本文のときは記事の先頭から何ページ目かを表す。文字領域のときは、/1 のみとなる。

各枠(ページ)の残りエリア用辞書

属性名	属性値の型	説明
/1 段目	辞書	1 段目の各エリアの配列
/2 段目	辞書	2段目の各エリアの配列
:		

各段のエリアの配列

「<1 つ目のエリアのエリア用配列> <2 つ目のエリアのエリア用配列> ...]

エリア用配列

[X座標値 Y座標値 幅 高さ]

X座標値、Y座標値、幅、高さは、単位がmmの単位付き数値文字列

サンプルコード

```
// 返り値の例
<<
 /1 << // 1つ目の枠
  /1 段目[] // 1 つ目の枠の 1 段目のエリア配列
  /2段目[] // 1つ目の枠の2段目のエリア配列
  /3 段目 // 1 つ目の枠の 3 段目のエリア配列
    「『10.0 mm" "10.0 mm" "20.0 mm" "12.5 mm" ] // 1つ目のエリアのエリア用配列
    [ "10.0 mm" "20.0 mm" "45.0 mm" "12.5 mm" ] // 2つ目のエリアのエリア用配列
  1
 >>
>>
// 1つ目の枠の1段目、1つ目のエリアの空領域サイズの固定枠を作成する
var bounds
"A000" @枠選択
@残りエリア /1 get /1 段目 get 0 get bounds @代入
/座標値 << /X座標 bounds @参照 0 get /Y座標 bounds @参照 1 get >>
/枠名称 "waku"
/矩形サイズ << /幅 bounds @参照 2 get /高さ bounds @参照 3 get >>
>> @固定枠作成
exit
```

検証者 memo

説明では「文字が途中までしか流れていないところの残りの矩形部分が」と記述されているが実際には文字が途中まで入っている部分は空き領域と判断されていないようである。

エリアが複数できる状態は固定枠などで行が分断されている場合など。

本文を対象にするとエリアが取得できない。

文字が読み込まれている領域を調べる

ver. 5.0.0.0 以降

文字が読み込まれている領域を調べたい場合には、**@読込サイズ**を使用することで読み 込まれている領域の情報を矩形のサイズとして取得することができます。

矩形のサイズは行数、行長、高さを配列で返します。このときの高さは行の高さと行間 の総和となります。

コマンド名:@読込サイズ

パラメータ:なし

返り値: 読込みサイズ情報配列(配列型)

読込みサイズ情報配列

[行数(整数型) 行長(単位付き数値文字列型 mm) 高さ(単位付き数値文字列型 mm)]

サンプルコード

@読込サイズ // [206 "35.6mm" "107.98mm"] がスタックのトップに積まれる

あふれている枠を拡張する

ver. 5.0.0.0 以降

あふれている枠を拡張するには@あふれ拡張を使用します。

このコマンドは選択した小組枠・表枠に対して編集機能の「あふれ拡張」を使用します。

コマンド名:@あふれ拡張

パラメータ:なし 返り値 : なし

●小組枠の場合 記事編集画面での「書式」→「あふれ拡張」と同じ

●表枠の場合 記事編集画面での「表組」→「あふれ拡張」と同じ

行間を調整して追込む

ver. 5.0.0.0 以降

あふれている枠内の文章を行間を調整して追込むには**@行間追込み**を使用します。 このコマンドは選択した小組枠、文字領域に対して、機能の「行間追込み」を行います。 これは記事編集画面の「書式」→「行間追込み」メニューを選択した場合と同じ処理に なります。

選択した小組枠、文字領域に対してのみ使用できます。

コマンド名:@行間追込み

パラメータ: なし 返り値 : なし

@行間追込みを標準行間の箇所(行間を調節していない場所)のみに適用したいという場合には**@標準行間追込み**が使用できます。

コマンド名:@標準行間追込み

パラメータ: なし 返り値 : なし

カラーパレットに色を追加する

ver. 5.0.0.0 以降

カラーパレットに色を追加するには**@色追加**を使用します。 このコマンドは指定する名称、指定する色成分の色をカラーパレットに追加します。

コマンド名:@色追加

パラメータ:色名称(文字列型) 色成分文字列(文字列型)

返り値 : なし

色成分文字列

●16 進表現

RGB : "#rrggbb" (rr, gg, bb は、0~255 の数値を 16 進表現したもの)

CMYK : "#ccmmyykk" (cc,mm,yy,kk は、0~100の数値を16進表現したもの)

●10 進表現(Ver5.0 以降) ※成分の値がゼロのチャネルは省略可能(桁揃えのゼロは省略可)

RGB : "RxGxBx" (x は、000~255)

CMYK : "CxMxYxKx" (x は、000~100)

サンプルコード

```
"黄色" "#00006400" @色追加 // 16 進表現
"黄色" "C000M000Y100K000" @色追加 // 10 進表現
"黄色" "Y100" @色追加 // 10 進表現(省略表現)
```

カラーパレットの色の成分を変更する

ver. 5.0.0.0 以降

カラーパレットに登録されている色の色成分を変更するには**@色成分変更**を使用します。 このコマンドは指定する名称の色の色成分を、指定する色成分に変更することができま す。

同名の色が複数存在する場合は、カラーパレット上で先頭の色が対象となります。

指定する名称の色が見つからなかった場合はエラーとなります。また、変更対象の色の 色成分のカラースペースが、指定する色成分のカラースペースと異なっている場合もエ ラーとなります。

コマンド名:@色成分変更

パラメータ:色名称(文字列型) 色成分文字列(文字列型)

返り値 :なし

色成分文字列

●16 進表現

RGB : "#rrggbb" (rr,gg,bb は、0~255 の数値を 16 進表現したもの)

CMYK : "#ccmmyykk" (cc,mm,yy,kk は、0~100 の数値を 16 進表現したもの)

●10 進表現(Ver5.0 以降) ※成分の値がゼロのチャネルは省略可能(桁揃えのゼロは省略可)

RGB : "RxGxBx" (x は、000~255)

CMYK : "CxMxYxKx" (x は、000~100)

サンプルコード

"黄色" "#00006400" @色成分変更 // 16 進表現

"黄色" "C000M000Y100K000" @色成分変更 // 10 進表現

"黄色" "Y100" @色成分変更 // 10 進表現(省略表現)

カラーパレットの色を削除する

ver. 5.0.0.0 以降

カラーパレットの色を削除するには**@色削除**を使用します。 このコマンドは指定する名称、色成分の色をカラーパレットから削除します。 指定する色がカラーパレット上に存在しない場合はエラーとなります。

コマンド名:@色削除

パラメータ:色名称(文字列型) 色成分文字列(文字列型)

返り値 : なし

色成分文字列

●16 進表現

RGB : "#rrggbb" (rr, gg, bb は、0~255 の数値を 16 進表現したもの)

CMYK : "#ccmmyykk" (cc,mm,yy,kk は、0~100 の数値を 16 進表現したもの)

●10 進表現(Ver5.0 以降)※成分の値がゼロのチャネルは省略可能(桁揃えのゼロは省略可)

RGB : "RxGxBx" (x は、000~255)

CMYK : "CxMxYxKx" (x は、000~100)

サンプルコード

"黄色" "#00106400" @色削除

オブジェクトの色を変更する

ver. 5.0.0.0 以降

記事内のオブジェクト(枠、図形、文字等)に設定されている色を変更するには **@色変更**を使用します。

変更対象の色および変更後の色の指定は色変更辞書で行います。

「/変換対象」で指定した部位に対して、「/変換元」で指定した色を「/変換先」の色に変更します。

「/変換元」「/変換先」に指定する色がカラーパレットに存在しない場合は、エラーとなります。

コマンド名:@色変更

パラメータ: 色変更辞書(辞書型)

返り値 :なし

色変更辞書

属性名	属性値の型	説明	備考
/変換元	文字列	文字列変換元の色名称を指定する。	必須
/変換先	文字列	文字列変換先の色名称を指定する。	必須
/変換対象	文字列	色の変換対象を指定する。	省略時
		"全て"、"文字"、"図形"、"枠"、"枠図形" のいずれか	"全て"

※V5.0以降は色名称にシステムカラーの指定が可能。

サンプルコード

<< /变換元 "黄色" /変換先 "赤" /変換対象 "枠" >> @色変更

// 環境設定でカラーパレット・グレイ色の設定値が「グレイ色として扱う」の場合

<< /変換元 "Gray 50%" /変換先 "黒" /変換対象 "全て" >> @色変更

// 環境設定でカラーパレット・グレイ色の設定値が「CMYK の K 版として扱う」の場合

<< /変換元 "K 50%" /変換先 "黒" /変換対象 "全て" >> @色変更

カラーパレットに色が存在するかチェックする

ver. 5.0.0.0 以降

指定の色がカラーパレットに存在しているかを確認するには2種類の方法があります。

● 色名称 + 色成分で調べる 色名称と色成分で指定された色の存在を確認するには**@色確認**を使用します。

コマンド名:@色確認

パラメータ: 名称文字列(文字列型) 色成分文字列(文字列型)

返り値 :真または偽(整数型)

色成分文字列

●16 進表現

RGB : "#rrggbb" (rr,gg,bb は、0~255の数値を 16 進表現したもの)

CMYK : "#ccmmyykk" (cc,mm,yy,kk は、0~100の数値を16進表現したもの)

●10 進表現(Ver5.0 以降)※成分の値がゼロのチャネルは省略可能(桁揃えのゼロは省略可)

RGB : "RxGxBx" (x は、000~255) CMYK : "CxMxYxKx" (x は、000~100)

サンプルコード

"黄色" "#00106400" @色確認

●色名称で調べる

色名称のみで指定された色の存在を確認したい場合には@色名称確認を使用します。

コマンド名:@色名称確認

パラメータ: 色名称(文字列型) 返り値: 真または偽(整数型)

サンプルコード

"黄色" @色名称確認

トリガーテキストを読み込む

ver. 5.0.0.0 以降

トリガーテキストの読み込みを行うには**@トリガー自動読込**を使用します。

パラメータに文字列を渡すと、その文字列をトリガーテキストとして自動読み込みする ことが出来ます。

コマンド名:@トリガー自動読込

パラメータ:トリガー自動読込み辞書(辞書型)または文字列型

返り値 : なし

トリガー自動読込み辞書

属性名	属性値の型	説明	備考
/ファイル名	文字列	トリガーのファイル名を指定する	必須
/種別	文字列	読込モードを指定する	必須
		"テキスト"、"トリガー"、"マクロ"のいずれか	

@トリガー自動読込は文字列の選択状態によって処理が変わります。

文字列が選択状態にあるときは、編集画面でのテキスト読込パネル上における「自動読込」の設定値は「テキスト」となりますが、文字列が選択されていない場合には「自動読込」の設定値は「レイアウト」となります。

「自動読込」の設定値が「テキスト」の状態で枠作成等のトリガーを読ませるとエラーとなります。

トリガーファイル名は、絶対パスか相対パスで指定できます。

相対パスで指定した場合には環境変数 EDITEXT からの相対パスになります。

また、@トリガー自動読込の実行時の読込規則ファイルを指定するには@読込規則設定を使用します。

読込規則はこのコマンド実行時にメモリ上に読み込まれて保持されるため、以降本コマンドが実行されるまで有効となります。

読込規則ファイル名は、絶対パスか相対パスで指定できます。

相対パスで指定した場合には環境変数 EDIRRLDIR からの相対パスになります。

コマンド名:@読込規則設定

パラメータ:読込規則ファイル名(文字列型)

返り値 : なし

サンプルコード

"IBM_SJIS2.rrl" @読込規則設定 // "IBM_SJIS2.rrl" を読み込み規則として設定

// ファイル名 T0020.txt のマクロを使ったトリガーファイルを自動読み込みする

<<

/ファイル名 "T0020.txt"

/種別 "マクロ"

>> @トリガー自動読込

exit

// 文字列をトリガーテキストとして自動読み込みする。

@文字列選択解除 // テキスト読込モードをレイアウトにする

"¥字下げ(3.50000mm,0.00000mm,0.00000mm,しない) NL TEST"

@トリガー自動読込

exit

トリガーテキストを取得する

ver. 5.0.0.0 以降

選択中の枠・領域のトリガーテキストをファイルに書き出すには**@トリガー書出**を使用します。

このコマンドの実行にはトリガー書出オプション定義ファイルを作成しておく必要があります。

コマンド名:@トリガー書出

パラメータ:トリガー書出辞書(辞書型)

返り値 : なし

トリガー書出辞書

属性名	属性値の型	説明	備考
/ファイル名		トリガーのファイル名を指定する 拡張子は不要 相対パス:環境変数 EDITEXT からの相対パス	必須
/オプション		トリガー書出オプション定義ファイルを指定する 相対パス:環境変数 EDI_ESL_TEXT からの相対パス	必須

サンプルコード

// オプション定義ファイル esl_gj.opt を使ってファイル名 T0020.TXT にトリガー書出を 行う

<<

/ファイル名 "T0020" // 拡張子は不要

/オプション "esl_gj.opt"

>> @トリガー書出

exit

選択している文字列のトリガーテキストを文字列 で取得する

ver. 5.0.0.0 以降

選択した文字列のトリガーを文字列型のオブジェクトとして取得するには**@トリガー取 得**を使用します。

コマンド名:@トリガー取得

パラメータ:なし

返り値 : トリガーコマンド(文字列型)

@トリガー取得実行時の設定は**@トリガー設定**で読み込んだオプション定義ファイルが使用されます。

このオプション定義ファイルは、**@トリガー書出**のトリガー書出オプション定義ファイルと同じものです。

オプション定義ファイル名は、絶対パスか相対パスで指定できます。

相対パスで指定した場合には環境変数 EDI_ESL_TEXT からの相対パスになります。

コマンド名:@トリガー設定

パラメータ:オプション定義ファイル名

返り値 : なし

@トリガー設定を実行していないときは、以下の設定でトリガーテキストが取得されます。

対象:選択

トリガーコマンド:なし

テキスト時外字:タグ付き

外字タグ: "¥j"

サンプルコード


"esl gi.opt" @トリガー設定

@トリガー取得

トリガー書出オプション定義ファイルを作成する

ver. 5.0.0.0 以降

EDIAN 編集画面の「テキスト書出」パネルから以下の方法で作成することができます。



「テキスト書出パネル」では設定できない追加定義

枠指定コマンド出力 = "する" / "しない" 「しない」を指定すると トリガーあり :枠指定コマンドを書き出さない トリガーなし :区切り+枠名称を書き出さない(区切り出力する場合のみ) ※省略値は「する」

サンプルコード

<オプション定義ファイルの表題> ## 対象="選択" 文字コード="Shift JIS" ページ形式="1ファイル" ページ指定="" トリガーコマンド="標準" 枠名称生成="しない" 小組枠毎の書出し="しない" 反射枠の書出し位置="混在" 欧文書体文字="半角" 改行文字="CR/LF" 組体裁情報="しない" 包含オブジェクトの書出し="する" 書出しレイヤ="255" ##「トリガーあり」タブ 書出し内容="全部" トリガー引数名="なし" 画像ファイル出力="する" 画像作成コマンド="しない" 基準点="用紙" 再配置可能="しない" 文字の単位="Q" 座標の単位="MM"

##「トリガーなし」タブ

区切り出力="しない" 枠名称書出="する" テキスト時外字="外字トリガー" 書出し規則="なし" 文章内データ="しない"

##「表組」タブ

セル中改行文字="" セル区切り="@td@" 行/列区切り="@tr@" 表の書出し順序="Z" トリガー保護="内容/区切り"

枠指定コマンド出力 = "する"

セルの選択をする

ver. 5.0.0.0 以降

選択中の表のセルを1つ選択するには@セル選択を使用します。

@セル選択にはパラメータの型によって、4つの選択方法があります。

整数型	セル順序(1 origin)
配列型	行と列(1 origin)
辞書型	セル属性辞書 ※セル順序で最初に一致したものが選択される
文字列型	セル中文字列 ※セル順序で最初に一致したものが選択される

セル属性、文字列による選択で一致するセルが複数あったときは、セル順序で最初に一致したものが選択されます。

セルの選択に成功した時は真(1)、失敗したときは偽(0)がスタックのトップに積まれ、表が選択されていないときはエラーとなります。

セル中文字列内の外字は¥jxxxx(16進数)で指定します。

また表の状態により、選択条件が変化します。

同名称の表枠が複数ある記事での選択条件

選択方法	選択条件
セル順序	先頭の表枠からの連番
行と列	先頭の表枠からの連番
セル属性	先頭の表枠から順に検索し条件にマッチしたセルを選択
文字列	先頭の表枠から順に検索し条件にマッチしたセルを選択

表頭が設定された表枠での選択条件

選択方法	選択条件
セル順序	表頭セルは対象外
行と列	表頭セルは対象外
セル属性	表頭セルも対象
文字列	表頭セルも対象

コマンド名:@セル選択

パラメータ:整数型、配列型、セル属性辞書(辞書型)、文字列型のいずれか

返り値 : 真または偽(整数型)

セル属性辞書

属性名	属性値の型	説明	備考
/レイヤ	整数	レイヤ番号を指定する	省略可
/文字レイヤ	整数	文字のレイヤ番号を指定する	省略可
/文字色	文字列	文字色を指定する	省略可*1)
/上線種	整数	上線の線種番号を指定する	省略可
/下線種	整数	下線の線種番号を指定する	省略可
/右線種	整数	右線の線種番号を指定する	省略可
/左線種	整数	左線の線種番号を指定する	省略可
/斜線種	整数	斜線の線種番号を指定する	省略可
/上線色	文字列	上線の色を指定する	省略可*1)
/下線色	文字列	下線の色を指定する	省略可*1)
/右線色	文字列	右線の色を指定する	省略可*1)
/左線色	文字列	左線の色を指定する	省略可*1)
/斜線色	文字列	斜線の色を指定する	省略可*1)
/面種	整数	面種の番号を指定する	省略可
/面色	文字列	面種の色を指定する	省略可*1)
/斜線形状	文字列	斜線の形状を指定する	省略可
		"なし"、"∕"、"∖"、"×" のいずれかを指定する。	
/あふれ処理	文字列	あふれ処理を指定する	省略可
		"しない"、"変形" のどちらかを指定する。	
/あふれ詰込	文字列	あふれ詰込を指定する	省略可
		"しない"、"表に従う"、"<詰込規則番号>" のいずれか	
		を指定する。<詰込規則番号>は任意の整数値	
/共通体裁	辞書	文字サイズ、書体等を指定する	省略可

^{*1)} 色の指定は、<色成分: 色名称> で指定する。 例) "C0M100Y100K0: 赤"

共通体裁辞書

属性名	属性値の型	説明	備考
/和文書体	文字列/整数	和文の書体を指定する	省略可*2)
/欧文書体	文字列/整数	欧文の書体を指定する	省略可*2)
/和文サイズ	文字列	和文の文字サイズを指定する (単位は、mm、pt、H、Q)	省略可
/欧文サイズ	文字列	欧文の文字サイズを指定する (単位は、mm、pt、H、Q)	省略可
/斜体	文字列	斜体を指定する "xx 度" または "なし" xx はー30~30	省略可
/変形	文字列	変形を指定する "長体 xx%"、"平体 xx%"、"なし" xx は 0~90	省略可
/文字間	文字列	文字間を指定する	省略可*3)
/行間	文字列	行間を指定する	省略可*3)
/かな詰め	文字列	かな詰めを指定する "あり"、"する"、"なし"、"しない" のいずれか	省略可
/行揃え	文字列	行揃えを指定する "上"、"中"、"下"、"行どり"(行取りも可)のいずれか	省略可
/字揃え	文字列	字揃えを指定する "左"、"中"、"右"、"字どり"(字取りも可)のいずれか	省略可

- *2) 文字列で書体名か整数値で書体番号のいずれかを指定する。
 - 例)文字列:"<C> A-OTF 新ゴ Pro L" 整数:9036
- *3) 値は、"<長さ>:<アキ/送り指定>"の形式を取る。

<長さ>: 単位付き数値文字列(単位は、mm,pt,H,Q)。

<アキ/送り指定>: "アキ"、"TOP"、"CENTER"、"BOTTOM"のいずれかで省略値は"アキ"となる。

文字間では、"BOTTOM"を指定するとエラーとなる。

例) "1.00pt:アキ"

サンプルコード

10 @セル選択 // セル順序 10 番目のセルを選択する

[23] @セル選択 // 2行目3列目のセルを選択する

<< /レイヤ3 >> @セル選択 // レイヤ番号3のセルを選択する

"あああ" @セル選択 // "あああ" をテキストとして持つセルを選択する

セル属性をコピー・ペーストする

ver. 5.0.0.0 以降

選択中のセルに対して、属性のコピー、ペーストを行には**@セル属性**を使用します。 コピー・ペーストの対象となる属性は、編集時の属性複写・属性貼込みと同じです。 セルが選択されていないときはエラーとなり、実行を中止します。

コマンド名:@セル属性

パラメータ: "コピー" または "ペースト"(文字列型)

返り値 :なし

サンプルコード

"コピー" @セル属性 // セル属性をコピー "ペースト" @セル属性 // セル属性をベースト

選択中のセルの文字列の選択する

ver. 5.0.0.0 以降

選択中のセル内の全テキストを選択状態にするには**0セル文字列選択**を使用します。 セルが選択されていないときはエラーとなり、実行を中止します。

コマンド名:@セル文字列選択

パラメータ: なし 返り値 : なし

サンプルコード

// 1番目のセルのトリガーテキストを取得する

"tbl O" @枠選択 // 名称が "tbl O" の枠を選択

1 @セル選択 // 1 番目のセルを選択

●セル文字列選択●トリガー取得

セルの選択位置を変更する

ver. 5.0.0.0 以降

セルの選択位置を現在の位置から相対指定で移動したい場合には**@セル選択移動**を使用します。

移動先にセルがないときには移動は失敗となり、コマンド実行による移動は行われません。移動に成功した時は真(1)、失敗したときは偽(0)がスタックのトップに積まれ、セルが選択されていないときはエラーとなり、実行を中止します。

コマンド名:@セル選択移動

パラメータ: "前行"、"次行"、"前列"、"次列" のいずれか(文字列型)

返り値 : 真または偽(整数型)

選択中のセルの位置情報を取得する

ver. 5.0.0.0 以降

@セル位置取得を使うと選択中のセルの位置が段(表)の先頭か、最後かといった位置情報を取得することが出来ます。

セルが選択されていないときはエラーとなり、実行を中止します。

コマンド名:@セル位置取得

パラメータ:なし

返り値 :""、"段頭"、"段末"、"段頭かつ段末"、"表頭"のいずれか(文字列型)

セルを結合する

ver. 5.0.0.0 以降

セルを結合するには@セル結合を使用します。

選択中のセルを基点に、横方向に結合するセル数と縦方向に結合するセル数を配列型で 指定します。このときのセル数は選択されているセル自身を含みます。

結合後は、結合されたセルが選択状態となり、結合対象のセル中の文字列がセルに入ります。

結合に成功すると真(1)が、失敗すると偽(0)がスタックのトップに積まれます。また、セルが選択されていないときはエラーとなり、実行を中止します。

失敗する条件は、記事編集時のセル結合と同じです。

コマンド名:@セル結合

パラメータ:縦構セル指定配列(配列型)

返り値 :真または偽(整数型)

縦横セル指定配列

[横方向のセル数(整数型)縦方向のセル数(整数型)]

セルの分割する

ver. 5.0.0.0 以降

選択中のセルを分割するには@セル分割を使用します。

@セル分割はパラメータの型によって、2つの分割方法が選択されます。

配列型	縦横共に均等に分割		
辞書型	セルの幅、	高さを指定して分割	(分割パネルの分割を指示するときと同じ分割)

分割後は、分割されたセル群の左上のセルが選択状態となり、分割前のセル中の文字列 が選択状態となるセルに入ります。

分割に成功すると真(1)が、失敗すると偽(0)がスタックのトップに積まれます。また、セルが選択されていないときはエラーとなり、実行を中止します。

失敗する条件は、記事編集時のセル分割と同じです。

コマンド名:@セル分割

パラメータ:縦横セル指定(配列型)またはセル分割辞書(辞書型)

返り値 :真または偽(整数型)

縦横セル指定配列

[横方向のセル数(整数型)縦方向のセル数(整数型)]

セル分割辞書

属性名	属性値の型	説明	備考
/縦分割	文字列	セル分割パネルに入力する分割指示文字列	省略可
/横分割	文字列	セル分割パネルに入力する分割指示文字列	省略可

分割指示文字列については、マニュアルのセル分割の項を参照してください。

選択中のセルの属性を変更する

ver. 5.0.0.0 以降

選択中のセルの属性を変更するには**@セル属性変更**を使用します。

コマンド名:@セル属性変更

パラメータ:セル属性辞書(辞書型)

返り値 : なし

セル属性辞書

属性名	属性値の型	説明	備考
/レイヤ	整数	レイヤ番号を指定する	省略可
/文字レイヤ	整数	文字のレイヤ番号を指定する	省略可
/文字色	文字列	文字色を指定する	省略可*1)
/上線種	整数	上線の線種番号を指定する	省略可
/下線種	整数	下線の線種番号を指定する	省略可
/右線種	整数	右線の線種番号を指定する	省略可
/左線種	整数	左線の線種番号を指定する	省略可
/斜線種	整数	斜線の線種番号を指定する	省略可
/上線色	文字列	上線の色を指定する	省略可*1)
/下線色	文字列	下線の色を指定する	省略可*1)
/右線色	文字列	右線の色を指定する	省略可*1)
/左線色	文字列	左線の色を指定する	省略可*1)
/斜線色	文字列	斜線の色を指定する	省略可*1)
/面種	整数	面種の番号を指定する	省略可
/面色	文字列	面種の色を指定する	省略可*1)
/斜線形状	文字列	斜線の形状を指定する	省略可
		"なし"、"∕"、"∖"、"×" のいずれかを指定する。	
/あふれ処理	文字列	あふれ処理を指定する	省略可
		"しない" "変形" のどちらかを指定する。	
/あふれ詰込	文字列	あふれ詰込を指定する	省略可
		"しない"、"表に従う"、"<詰込規則番号>" のいずれか	
		を指定する。<詰込規則番号>は任意の整数値	
/共通体裁	辞書	文字サイズ、書体等を指定する	省略可

^{*1)} 色の指定は、<色成分:色名称> で指定する。 例) "C0M100Y100K0:赤"

共通体裁辞書

属性名	属性値の型	説明	備考
/和文書体	文字列/整数	和文の書体を指定する	省略可*2)
/欧文書体	文字列/整数	欧文の書体を指定する	省略可*2)
/和文サイズ	文字列	和文の文字サイズを指定する (単位は、mm、pt、H、Q)	省略可
/欧文サイズ	文字列	欧文の文字サイズを指定する (単位は、mm、pt、H、Q)	省略可
/斜体	文字列	斜体を指定する "xx 度" または "なし" xx はー30~30	省略可
/変形	文字列	変形を指定する "長体 xx%"、"平体 xx%"、"なし" xx は 0~90	省略可
/文字間	文字列	文字間を指定する	省略可*3)
/行間	文字列	行間を指定する	省略可*3)
/かな詰め	文字列	かな詰めを指定する "あり"、"する"、"なし"、"しない" のいずれか	省略可
/行揃え	文字列	行揃えを指定する "上"、"中"、"下"、"行どり"(行取りも可)のいずれか	省略可
/字揃え	文字列	字揃えを指定する "左"、"中"、"右"、"字どり"(字取りも可)のいずれか	省略可

*2) 文字列で書体名か整数値で書体番号のいずれかを指定する。

例) 文字列: "<C> A-OTF 新ゴ Pro L" 整数: 9036

*3) 値は、"<長さ>:<アキ/送り指定>"の形式を取る。

<長さ>: 単位付き数値文字列(単位は、mm,pt,H,Q)。

<アキ/送り指定>: "アキ"、"TOP"、"CENTER"、"BOTTOM" のいずれかで省略値は"アキ"となる。

文字間では、"BOTTOM"を指定するとエラーとなる。

例) "1.00pt:アキ"

サンプルコード

// 表枠を選択しセル順序 1 番目のセルの文字色を変更する

"HYO" @枠選択

1 @セル選択

<<

/文字色 "Gray100:黒" // システム環境によっては "K100:黒"

>> @セル属性変更

exit

選択中の表のセルサイズを調整する

ver. 5.0.0.0 以降

選択中の表に対して、編集機能の「セルサイズ調整」を適用するには**@セルサイズ調整**を使用します。

コマンド名:@セルサイズ調整

パラメータ: なし 返り値 : なし

選択中のセルから条件にあった次のセルを選択する

ver. 5.0.0.0 以降

選択中のセルの次の位置から指定した条件に合致するセルを、選択するには**@次セル選** 状を使用します。

セルの選択に成功した時は真(1)、失敗したときは偽(0)がスタックのトップに積まれます。表が選択されていないときはエラーとなり、実行を中止します。

条件には「セル属性」(辞書型)または「セル中文字列」(文字列型)を選択することができます。

辞書型	セル属性を条件に検索する		
文字列型	セル中文字列を条件に検索する		
	セル中文字列内の外字は¥jxxxx(16 進数)で指定する		

※@セル選択で使用できる「セル順序」と「行と列」による検索はできません。

同名称の表枠が複数ある記事での選択条件

選択方法	選択条件
セル順序	機能無し(エラーとなる)
行と列	機能無し(エラーとなる)
セル属性	先頭の表枠から順に検索し条件にマッチしたセルを選択
文字列	先頭の表枠から順に検索し条件にマッチしたセルを選択

表頭が設定された表枠での選択条件

選択方法	選択条件
セル順序	機能無し(エラーとなる)
行と列	機能無し (エラーとなる)
セル属性	表頭セルも対象
文字列	表頭セルも対象

コマンド名:@次セル選択

パラメータ: セル属性辞書(辞書型)またはセル中文字列(文字列型)

返り値 : 真または偽(整数型)

セル属性辞書

属性名	属性値の型	説明	備考
/レイヤ	整数	レイヤ番号を指定する	省略可
/文字レイヤ	整数	文字のレイヤ番号を指定する	省略可
/文字色	文字列	文字色を指定する	省略可*1)
/上線種	整数	上線の線種番号を指定する	省略可
/下線種	整数	下線の線種番号を指定する	省略可
/右線種	整数	右線の線種番号を指定する	省略可
/左線種	整数	左線の線種番号を指定する	省略可
/斜線種	整数	斜線の線種番号を指定する	省略可
/上線色	文字列	上線の色を指定する	省略可*1)
/下線色	文字列	下線の色を指定する	省略可*1)
/右線色	文字列	右線の色を指定する	省略可*1)
/左線色	文字列	左線の色を指定する	省略可*1)
/斜線色	文字列	斜線の色を指定する	省略可*1)
/面種	整数	面種の番号を指定する	省略可
/面色	文字列	面種の色を指定する	省略可*1)
/斜線形状	文字列	斜線の形状を指定する	省略可
		"なし"、"∕"、"∖"、"×" のいずれかを指定する。	
/あふれ処理	文字列	あふれ処理を指定する	省略可
		"しない"、"変形" のどちらかを指定する。	
/あふれ詰込	文字列	あふれ詰込を指定する	省略可
		"しない"、"表に従う"、"<詰込規則番号>" のいずれか	
		を指定する。<詰込規則番号>は任意の整数値	
/共通体裁	辞書	文字サイズ、書体等を指定する	省略可

^{*1)} 色の指定は、<色成分: 色名称> で指定する。 例) "C0M100Y100K0: 赤"

共通体裁辞書

属性名	属性値の型	説明	備考
/和文書体	文字列/整数	和文の書体を指定する	省略可*2)
/欧文書体	文字列/整数	欧文の書体を指定する	省略可*2)
/和文サイズ	文字列	和文の文字サイズを指定する (単位は、mm、pt、H、Q)	省略可
/欧文サイズ	文字列	欧文の文字サイズを指定する (単位は、mm、pt、H、Q)	省略可
/斜体	文字列	斜体を指定する "xx 度" または "なし" xx は-30~30	省略可

/変形	文字列	変形を指定する "長体 xx%"、"平体 xx%"、"なし" xx は 0~90	省略可
/文字間	文字列	文字間を指定する	省略可*3)
/行間	文字列	行間を指定する	省略可*3)
/かな詰め	文字列	かな詰めを指定する "あり"、"する"、"なし"、"しない" のいずれか	省略可
/行揃え	文字列	行揃えを指定する "上"、"中"、"下"、"行どり"(行取りも可)のいずれか	省略可
/字揃え	文字列	字揃えを指定する "左"、"中"、"右"、"字どり"(字取りも可)のいずれか	省略可

- *2) 文字列で書体名か整数値で書体番号のいずれかを指定する。
 - 例) 文字列:"<C> A-OTF 新ゴ Pro L" 整数:9036
- *3) 値は、"<長さ>:<アキ/送り指定>"の形式を取る。

<長さ>: 単位付き数値文字列(単位は、mm,pt,H,Q)。

<アキ/送り指定>: "アキ"、"TOP"、"CENTER"、"BOTTOM" のいずれかで省略値は "アキ" となる。文字間では、"BOTTOM" を指定するとエラーとなる。

例) "1.00pt:アキ"

サンプルコード

// 表枠を選択し空のセルの属性を変更する

"HYO" @枠選択

(""@次セル選択)

<<

/面種 1

/面色 "Gray10: Gray10%" // 環境設定によっては "K10: K10%"

/斜線形状 "/"

>> @セル属性変更

loop

exit

選択中の文字列を削除する

ver. 5.0.0.0 以降

選択中の文字列を削除したい場合には**@文字列削除**を使用します。 文字列が選択されていないときはエラーとなり、実行を中止します。

コマンド名:@文字列削除

パラメータ: なし 返り値 : なし

文字列の選択状態を調べる

ver. 5.0.0.0 以降

文字列が選択されているかどうかを調べるには**@文字列選択状態**を使用します。

文字列が選択されているか、文字カーソルが立っているときは真(1)、そうでないときは偽(0)がスタックのトップに積まれます。

コマンド名:@文字列選択状態

パラメータ:なし

返り値 : 真または偽(整数型)

文字列の選択を解除する

ver. 5.0.0.0 以降

文字列の選択を解除するには@文字列選択解除を使用します。

このコマンドは文字列型のパラメータで**®トリガー自動読込**を実行するときの状態を制御するときに使用されます。

文字列が選択状態にあるときは、テキスト読込パネルの「自動読込」の設定値は「テキスト」となりますが、本コマンドによって、文字列が選択状態でなくなったときは、設定値が「レイアウト」となります。

「自動読込」の設定値が「テキスト」の状態で枠作成等のトリガーを読ませると、エラーとなります。

これは編集時の動作と同じです。

コマンド名:@文字列選択解除

パラメータ: なし 返り値 : なし

単位付き数値文字列の単位を変換する

ver. 5.0.0.0 以降

単位付き数値文字列の単位を変換するには**@単位変換**を使用します。 指定単位には、"mm"、"pt"、"ees" の3つが指定出来ます。

コマンド名:@単位変換

パラメータ:単位付き数値文字列(文字列型)単位(文字列型)

返り値 :変換済みの単位付き数値文字列(文字列型)

サンプルコード

"**72pt" "mm" @単位変換** // "25.4mm" がスタックのトップに積まれる

記事名を取得する

ver. 5.0.0.0 以降

編集スクリプトが実行されている記事の記事名を取得するには**@記事名**を使用します。

コマンド名:@記事名 パラメータ:なし

返り値:記事名(文字列型)

文書名を取得する

ver. 5.0.0.0 以降

編集スクリプトが実行されている記事の文書名を取得するには@文書名を使用します。

コマンド名:@文書名 パラメータ:なし

返り値 : 文書名(文字列型)

文章内データ一覧を取得する

ver. 5.0.0.0 以降

記事内にある文章内データ情報の一覧を取得するには**@文章内データ一覧**を使用します。 文章内データの一覧は、配列型でスタックのトップに積まれ、その要素1つが1つの文 書内データ情報となります。文書内データ情報自身も配列型で、用途番号(整数型)、設 定文字列(文字列型)、ノンブル(整数型)の順でデータが格納されます。

文章内データの順序は記事編集時の文章内データの検索順序と同じです。

このコマンドの実行後には、枠の選択は解除されます。

コマンド名:@文章内データ一覧

パラメータ:なし

返り値 : 文章内データ一覧(配列型)

文章内データ一覧

[文章内データ情報配列(配列型) 文章内データ情報配列(配列型) ……]

文章内データ情報配列

「用途番号(整数型) 設定文字列(文字列型) ノンブル(整数型)]

サンプルコード

```
// 返り値の例
[
[ 2 "文字列 1" 20 ]
[ 3 "文字列 2" 21 ]
]

// 文章内データ一覧サンプル
var path
"/@D/data/" path @代入 // 出力フォルダを変数に代入

// ファイル名作成
path @参照 @記事名 strcat ".txt" strcat path @代入 // path + "記事名" + ".txt"
```

```
// 文章内データ一覧を取得、変数 list に代入し変数 num に文章内データの数を代入する
var list
var num
var n
var fp
@文章内データ一覧 dup list @代入 length num @代入
// ループカウンタ&一覧配列のインデックス用変数(n)の初期化
0 n @代入
// 書き込み用ファイル (path) のオープンと、ファイルオブジェクトを変数 (fp) に代入
path @参照 open fp @代入
( n @参照 num @参照 < )
 list @参照 n @参照 get dup dup // 文章内データ一覧の配列から n 番目の文書内データを取り出し
 O get tostring " " strcat fp @参照 putstring // 用途番号+" " の書き出し
 1 get
           " " strcat fp @参照 putstring // 設定文字列+" " の書き出し
                  fp @参照 putstring // ノンブル+" " の書き出し
 2 get tostring
                  fp @参照 putstring // 改行の書き出し
 NL
 n @参照 1 add n @代入 // カウンターをインクリメント
loop
fp @参照 close // ファイルを閉じる
exit
```

差込情報の一覧を取得する

ver. 5.0.0.0 以降

記事の差込情報の一覧を取得するには**@差込一覧**を使用します。

差込の一覧は、配列型でスタックのトップに積まれ、その要素1つが1つの差込情報となります。

差込情報の順序は、記事編集時の文章内データ(差込)の検索順序と同じです。 差込情報自身も配列型で、その要素は以下の順で入っています。

差込情報配列

開始ノンブル	整数型	
終了ノンブル	整数型	
枠種別	文字列型 (一覧パネルと同じ文字列)	
枠名称	文字列型(小組枠、文字領域、表以外は、長さゼロの文字列)	
差込 ID	文字列型	
差込文字列	文字列型	

差込文字列の書式(トリガー等の出力形式)は、**@トリガー取得**コマンドと同じです。 共通ページの開始/終了ページは、負の値となります。

共通ページが「単」の場合 -1、見開き記事と共通ページが「複」の場合、偶数ページが-1、奇数ページが-2となります。

差込情報内のアクセスは、先頭から何番目として行う必要があります。

最後の要素が差込文字列として処理していると、追加されたときに最後の要素は差込文字列ではなくなるためです。

コマンド名:@差込一覧

パラメータ:なし

返り値 : 差込情報の一覧(配列型)

サンプルコード

// 差込一覧をファイルに出力するサンプル

"オプションファイル名: esl gi.opt". // コンソールにオプションファイル名を表示

"esl_gj.opt" @トリガー設定

```
// 出力フォルダを変数に代入
var path
"/@D/EDIAN10A/USRDATA/editext/esl_out/" path @代入
// ファイル名作成 → path + "記事名" + ".txt"
path @参照 @記事名 strcat ".txt" strcat path @代入
// 差込一覧を取得、変数 list に代入し変数 num に文章内データの数を代入する
var list
var num
@差込一覧 dup list @代入 length num @代入
var n
var fp
On @代入 // ループカウンタ&一覧配列のインデックス用変数(n)の初期化
// 書き込み用ファイル(path)のオープンと、ファイルオブジェクトを変数(fp)に代入
path @参照 open fp @代入
( n @参照 num @参照 < )
 // 差込データ一覧の配列から n 番目の文書内データを取り出し、そのコピーを5つ作成しておく
 list @参照 n @参照 get dup dup dup dup
 O get tostring " " strcat fp @参照 putstring // 開始ノンブル+" " の書き出し
 1 get tostring " " strcat fp @参照 putstring // 終了ノンブル+" " の書き出し
            " " strcat fp @参照 putstring // 枠種別 の書き出し
 2 get
            " " strcat fp @参照 putstring // 枠名称 の書き出し
 3 get
 4 get
            " " strcat fp @参照 putstring // 設定文字列 の書き出し
                   fp @参照 putstring // 文字列 の書き出し
 5 get
 NL
                   fp @参照 putstring // 改行の書き出し
 n @参照 1 add n @代入
loop
fp @参照 close
exit
```

ノンブルからページ番号へ変換する

ver. 5.0.0.0 以降

ノンブルから、そのページのページ番号を取得するには**@ページ番号へ**を使用します。 文章内データ一覧・差込一覧で取得したノンブルのページへ移動する際などに使用しま す。

コマンド名: @ページ番号へ パラメータ: ノンブル(整数型) 返り値: ページ番号(整数型)

ページ番号からノンブルへ変換する

ver. 5.0.0.0 以降

ページ番号から、そのページのノンブルを取得するには**@ノンブルへ**を使用します。 パラメータに負の値を与えたときは、その値が返り値となります。

コマンド名:@ノンブルへ

パラメータ:ページ番号(整数型) 返り値 : ノンブル(整数型)

環境変数の値の取得する

ver. 5.0.0.0 以降

環境変数の値を取得する場合には@環境変数を使用します。

指定された環境変数の値がスタックのトップに積まれます。

環境変数が存在しなかったときは、空の文字列型オブジェクトが積まれます。

コマンド名:@環境変数

パラメータ:環境変数名(文字列型) 返り値:環境変数の値(文字列型)

基本環境変数名一覧

環境変数名	意味	値(スタンドアロン構成の例)
EDIGAZOU	標準画像フォルダ	D:¥EDIAN10A¥USRDATA¥edigazou
EDITEXT	標準テキストフォルダ	D:¥EDIAN10A¥USRDATA¥editext
EDIPSDIR	標準 PS 出力フォルダ	D:¥EDIAN10A¥USRDATA¥psout
EDI_ESL_TEXT	標準スクリプトフォルダ	D:¥EDIAN10A¥EDIDATA¥ediesl
EDIBAT	標準一括処理フォルダ	D:¥EDIAN10A¥SYSDATA¥edibatch

サンプルコード

// 環境変数からパスを取得してファイルパスを作成する

var FILEPATH

"EDITEXT" @環境変数

"/例 1.txt" strcat FILEPATH @代入

FILEPATH @参照 .

exit

// 環境変数の一覧をコンソールに表示する

"環境変数一覧".

"EDIPSDIR"dup ":" strcat swap @環境変数 strcat ."EDI_ESL_TEXT"dup ":" strcat swap @環境変数 strcat ."EDIROOT"dup ":" strcat swap @環境変数 strcat ."EDIDATADIR"dup ":" strcat swap @環境変数 strcat ."EDILIB"dup ":" strcat swap @環境変数 strcat ."EDIGAIJI"dup ":" strcat swap @環境変数 strcat .

```
"FDITOMBODIR"
                         dup ": " strcat swap @環境変数 strcat.
"FDIBAT"
                         dup ":" strcat swap @環境変数 strcat.
"EDIMATH"
                         dup ":" strcat swap @環境変数 strcat .
"FDIHYOR"
                         dup ":" strcat swap @環境変数 strcat .
"EDITEXT"
                         dup ":" strcat swap @環境変数 strcat .
"EDITEXT X1"
                         dup ":" strcat swap @環境変数 strcat .
"EDITEXT X2"
                         dup ":" strcat swap @環境変数 strcat.
"EDITEXT X3"
                         dup ":" strcat swap @環境変数 strcat .
                         dup ":" strcat swap @環境変数 strcat.
"EDITEXT X4"
"EDIRRLDIR"
                         dup ":" strcat swap @環境変数 strcat .
"EDIGAZOU"
                         dup ":" strcat swap @環境変数 strcat.
"EDIGAZOU X1"
                         dup ":" strcat swap @環境変数 strcat .
"EDIGAZOU X2"
                         dup ":" strcat swap @環境変数 strcat.
"EDIGAZOU X3"
                         dup ":" strcat swap @環境変数 strcat.
"EDIGAZOU X4"
                         dup ":" strcat swap @環境変数 strcat .
"EDIGAZOU X5"
                         dup ":" strcat swap @環境変数 strcat.
"EDIPARTS"
                         dup ":" strcat swap @環境変数 strcat.
"EDIMG"
                         dup ":" strcat swap @環境変数 strcat .
"EDIILL"
                         dup ":" strcat swap @環境変数 strcat .
"EDIPRTENVDIR"
                         dup ":" strcat swap @環境変数 strcat .
"GJED PFB READ FNAME" dup ":" strcat swap @環境変数 strcat.
"EDIPSF"
                         dup ":" strcat swap @環境変数 strcat.
"EDIDATADIR"
                         dup ":" strcat swap @環境変数 strcat.
"EDIBASEDIR"
                         dup ":" strcat swap @環境変数 strcat.
exit
```

118

警告パネルにメッセージを表示させる

ver. 5.0.0.0 以降

EdianWingの警告パネルに警告メッセージを表示させるには**@警告パネル**を使用します。

コマンド名:@警告パネル

パラメータ:警告メッセージ(文字列型)

返り値 :なし

サンプルコード

//「エラーです」を警告パネルに表示する

"エラーです"@警告パネル

警告パネル



確認パネルにメッセージを表示させる

ver. 5.0.0.0 以降

EdianWing の確認パネルに通知メッセージを表示させるには@確認パネルを使用します。

コマンド名:@確認パネル

パラメータ:通知メッセージ(文字列型)

返り値 : なし

サンプルコード

//「通知です」を確認パネルに表示する

"通知です"@確認パネル

確認パネル



選択パネルを使用する

ver. 5.0.0.0 以降

スクリプトの最中にユーザー判断による分岐を設定したい場合には、**@確認パネル**で選択パネルを使用することができます。

選択パネルはメッセージと「はい」「いいえ」のボタンを表示します。ユーザーが「はい」を選択した場合には真(1)が、「いいえ」を選択した場合には偽(0)が、スタックのトップに積まれます。

コマンド名:@選択パネル

パラメータ:メッセージ(文字列型) 返り値: 真または偽(整数型)

サンプルコード

//「選択して下さい」を選択パネルに表示して、選択結果をコンソールに表示する

"選択して下さい"@選択パネル if

"「はい」が選択されました".

else

"「いいえ」が選択されました".

endif

選択パネル



選択中の枠の枠名称を取得する

ver. 5.0.0.0 以降

選択中の枠の枠名称を取得するには@枠名称取得を使用します。

このコマンドは記事編集画面で枠を選択後、編集スクリプトを実行する用途を想定しています。

1つの枠を選択したときは返り値が文字列となり、複数の枠を選択したときは返り値が 配列となり、その要素が枠名称となります。

枠が選択されていないときはエラーとなります。

枠名称がないとき、またはその他のエラー時は空の文字列型オブジェクトがスタックのトップに積まれます。

コマンド名:@枠名称取得

パラメータ:なし

返り値 : 枠名称(文字列型または配列型)

サンプルコード

// 複数選択時「複数選択しています」の確認パネルを出す

@ 枠 名 称 取 得 object " 配 列 " == if

"複数選択しています"@確認パネル

endif

EdinWing 編集スクリプト CookBook